

SAS/ETS[®] 14.2 User's Guide

Working with Time Series Data

This document is an individual chapter from *SAS/ETS® 14.2 User's Guide*.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc. 2016. *SAS/ETS® 14.2 User's Guide*. Cary, NC: SAS Institute Inc.

SAS/ETS® 14.2 User's Guide

Copyright © 2016, SAS Institute Inc., Cary, NC, USA

All Rights Reserved. Produced in the United States of America.

For a hard-copy book: No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

For a web download or e-book: Your use of this publication shall be governed by the terms established by the vendor at the time you acquire this publication.

The scanning, uploading, and distribution of this book via the Internet or any other means without the permission of the publisher is illegal and punishable by law. Please purchase only authorized electronic editions and do not participate in or encourage electronic piracy of copyrighted materials. Your support of others' rights is appreciated.

U.S. Government License Rights; Restricted Rights: The Software and its documentation is commercial computer software developed at private expense and is provided with RESTRICTED RIGHTS to the United States Government. Use, duplication, or disclosure of the Software by the United States Government is subject to the license terms of this Agreement pursuant to, as applicable, FAR 12.212, DFAR 227.7202-1(a), DFAR 227.7202-3(a), and DFAR 227.7202-4, and, to the extent required under U.S. federal law, the minimum restricted rights as set out in FAR 52.227-19 (DEC 2007). If FAR 52.227-19 is applicable, this provision serves as notice under clause (c) thereof and no other notice is required to be affixed to the Software or documentation. The Government's rights in Software and documentation shall be only those set forth in this Agreement.

SAS Institute Inc., SAS Campus Drive, Cary, NC 27513-2414

November 2016

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

SAS software may be provided with certain third-party software, including but not limited to open-source software, which is licensed under its applicable third-party software license agreement. For license information about third-party software distributed with SAS software, refer to <http://support.sas.com/thirdpartylicenses>.

Chapter 3

Working with Time Series Data

Contents

Overview	66
Time Series and SAS Data Sets	67
Introduction	67
Reading a Simple Time Series	68
Dating Observations	69
SAS Date, Datetime, and Time Values	69
Reading Date and Datetime Values with Informats	71
Formatting Date and Datetime Values	71
The Variables DATE and DATETIME	73
Sorting by Time	73
Subsetting Data and Selecting Observations	74
Subsetting SAS Data Sets	74
Using the WHERE Statement with SAS Procedures	75
Using SAS Data Set Options	75
Storing Time Series in a SAS Data Set	76
Standard Form of a Time Series Data Set	77
Several Series with Different Ranges	78
Missing Values and Omitted Observations	78
Cross-Sectional Dimensions and BY Groups	79
Interleaved Time Series	81
Output Data Sets of SAS/ETS Procedures	83
Time Series Periodicity and Time Intervals	84
Specifying Time Intervals	85
Using Intervals with SAS/ETS Procedures	86
Time Intervals, the Time Series Forecasting System, and the Time Series Viewer	86
Plotting Time Series	86
Using the Time Series Viewer	87
Using PROC SGPLOT	87
Using PROC GPLOT	93
Calendar and Time Functions	94
Computing Dates from Calendar Variables	94
Computing Calendar Variables from Dates	95
Converting between Date, Datetime, and Time Values	95
Computing Datetime Values	95
Computing Calendar and Time Variables	96

Interval Functions INTNX and INTCK	96
Incrementing Dates by Intervals	97
Alignment of SAS Dates	98
Computing the Width of a Time Interval	99
Computing the Ceiling of an Interval	100
Counting Time Intervals	100
Checking Data Periodicity	101
Filling In Omitted Observations in a Time Series Data Set	101
Using Interval Functions for Calendar Calculations	102
Lags, Leads, Differences, and Summations	102
The LAG and DIF Functions	103
Multiperiod Lags and Higher-Order Differencing	106
Percent Change Calculations	107
Leading Series	109
Summing Series	110
Transforming Time Series	111
Log Transformation	111
Other Transformations	113
The EXPAND Procedure and Data Transformations	114
Manipulating Time Series Data Sets	114
Splitting and Merging Data Sets	114
Transposing Data Sets	115
Time Series Interpolation	119
Interpolating Missing Values	119
Interpolating to a Higher or Lower Frequency	120
Interpolating between Stocks and Flows, Levels and Rates	120
Reading Time Series Data	121
Reading a Simple List of Values	121
Reading Fully Described Time Series in Transposed Form	122

Overview

This chapter discusses working with time series data in the SAS System. The following topics are included:

- dating time series and working with SAS date and datetime values
- subsetting data and selecting observations
- storing time series data in SAS data sets
- specifying time series periodicity and time intervals
- plotting time series

- using calendar and time interval functions
- computing lags and other functions across time
- transforming time series
- transposing time series data sets
- interpolating time series
- reading time series data recorded in different ways

In general, this chapter focuses on using features of the SAS programming language and not on features of SAS/ETS software. However, since SAS/ETS procedures are used to analyze time series, understanding how to use the SAS programming language to work with time series data is important for the effective use of SAS/ETS software.

You do not need to read this chapter to use SAS/ETS procedures. If you are already familiar with SAS programming you might want to skip this chapter, or you can refer to sections of this chapter for help on specific time series data processing questions.

Time Series and SAS Data Sets

Introduction

To analyze data with the SAS System, data values must be stored in a SAS data set. A SAS data set is a matrix (or table) of data values organized into variables and observations.

The *variables* in a SAS data set label the columns of the data matrix, and the *observations* in a SAS data set are the rows of the data matrix. You can also think of a SAS data set as a kind of file, with the observations representing records in the file and the variables representing fields in the records. (For more information about SAS data sets, see *SAS Language Reference: Concepts*.)

Usually, each observation represents the measurement of one or more variables for the individual subject or item observed. Often, the values of some of the variables in the data set are used to identify the individual subjects or items that the observations measure. These identifying variables are referred to as *ID variables*.

For many kinds of statistical analysis, only relationships among the variables are of interest, and the identity of the observations does not matter. ID variables might not be relevant in such a case.

However, for time series data the identity and order of the observations are crucial. A time series is a set of observations made at a succession of equally spaced points in time.

For example, if the data are monthly sales of a company's product, the variable measured is sales of the product and the unit observed is the operation of the company during each month. These observations can be identified by year and month. If the data are quarterly gross national product, the variable measured is final goods production and the unit observed is the economy during each quarter. These observations can be identified by year and quarter.

For time series data, the observations are identified and related to each other by their position in time. Since SAS does not assume any particular structure to the observations in a SAS data set, there are some special considerations needed when storing time series in a SAS data set.

The main considerations are how to associate dates with the observations and how to structure the data set so that SAS/ETS procedures and other SAS procedures recognize the observations of the data set as constituting time series. These issues are discussed in the following sections.

Reading a Simple Time Series

Time series data can be recorded in many different ways. The section “[Reading Time Series Data](#)” on page 121 discusses some of the possibilities. The example that follows shows a simple case.

The following SAS statements read monthly values of the U.S. Consumer Price Index (CPI) for June 1990 through July 1991. The data set USCPI is shown in [Figure 3.1](#).

```
data uscpi;
    input year month cpi;
datalines;
1990  6 129.9
1990  7 130.4
1990  8 131.6

... more lines ...

proc print data=uscpi;
run;
```

Figure 3.1 Time Series Data

Obs	year	month	cpi
1	1990	6	129.9
2	1990	7	130.4
3	1990	8	131.6
4	1990	9	132.7
5	1990	10	133.5
6	1990	11	133.8
7	1990	12	133.8
8	1991	1	134.6
9	1991	2	134.8
10	1991	3	135.0
11	1991	4	135.2
12	1991	5	135.6
13	1991	6	136.0
14	1991	7	136.2

When a time series is stored in the manner shown by this example, the terms *series* and *variable* can be used interchangeably. There is one observation per row and one series/variable per column.

Dating Observations

The SAS System supports special date, datetime, and time values, which make it easy to represent dates, perform calendar calculations, and identify the time period of observations in a data set.

The preceding example uses the ID variables YEAR and MONTH to identify the time periods of the observations. For a quarterly data set, you might use YEAR and QTR as ID variables. A daily data set might have the ID variables YEAR, MONTH, and DAY. Clearly, it would be more convenient to have a single ID variable that could be used to identify the time period of observations, regardless of their frequency.

The following section, “[SAS Date, Datetime, and Time Values](#)” on page 69, discusses how the SAS System represents dates and times internally and how to specify date, datetime, and time values in a SAS program. The section “[Reading Date and Datetime Values with Informats](#)” on page 71 discusses how to read in date and time values from data records and how to control the display of date and datetime values in SAS output. Later sections discuss other issues concerning date and datetime values, specifying time intervals, data periodicity, and calendar calculations.

SAS date and datetime values and the other features discussed in the following sections are also described in *SAS Language Reference: Dictionary*. Reference documentation on these features is also provided in Chapter 4, “[Date Intervals, Formats, and Functions](#).”

SAS Date, Datetime, and Time Values

SAS Date Values

SAS software represents dates as the number of days since a reference date. The reference date, or date zero, used for SAS date values is 1 January 1960. For example, 3 February 1960 is represented by SAS as 33. The SAS date for 17 October 1991 is 11612.

SAS software correctly represents dates from the year 1582 to the year 20,000.

Dates represented in this way are called *SAS date values*. Any numeric variable in a SAS data set whose values represent dates in this way is called a *SAS date variable*.

Representing dates as the number of days from a reference date makes it easy for the computer to store them and perform calendar calculations, but these numbers are not meaningful to users. However, you never have to use SAS date values directly, since SAS automatically converts between this internal representation and ordinary ways of expressing dates, provided that you indicate the format with which you want the date values to be displayed. (Formatting of date values is explained in the section “[Formatting Date and Datetime Values](#)” on page 71.)

Century of Dates Represented with Two-Digit Year Values

SAS software informats, functions, and formats can process dates that are represented with two-digit year values. The century assumed for a two-digit year value can be controlled with the YEARCUTOFF= option in the OPTIONS statement. The YEARCUTOFF= system option controls how dates with two-digit year values are interpreted by specifying the first year of a 100-year span. The default value for the YEARCUTOFF= option is 1920. Thus by default the year ‘17’ is interpreted as 2017, while the year ‘25’ is interpreted as 1925. (For more information about the YEARCUTOFF= option, see *SAS Language Reference: Dictionary*.)

SAS Date Constants

SAS date values are written in a SAS program by placing the dates in single quotes followed by a D. The date is represented by the day of the month, the three letter abbreviation of the month name, and the year.

For example, SAS reads the value '17OCT1991'D the same as 11612, the SAS date value for 17 October 1991. Thus, the following SAS statements print DATE=11612:

```
data _null_;
  date = '17oct1991'd;
  put date=;
run;
```

The year value can be given with two or four digits, so '17OCT91'D is the same as '17OCT1991'D.

SAS Datetime Values and Datetime Constants

To represent both the time of day and the date, SAS uses *datetime values*. SAS datetime values represent the date and time as the number of seconds the time is from a reference time. The reference time, or time zero, used for SAS datetime values is midnight, 1 January 1960. Thus, for example, the SAS datetime value for 17 October 1991 at 2:45 in the afternoon is 1003329900.

To specify datetime constants in a SAS program, write the date and time in single quotes followed by DT. To write the date and time in a SAS datetime constant, write the date part using the same syntax as for date constants, and follow the date part with the hours, the minutes, and the seconds, separating the parts with colons. The seconds are optional.

For example, in a SAS program you would write 17 October 1991 at 2:45 in the afternoon as '17OCT91:14:45'DT. SAS reads this as 1003329900. Table 3.1 shows some other examples of datetime constants.

Table 3.1 Examples of Datetime Constants

Datetime Constant	Time
'17OCT1991:14:45:32'DT	32 seconds past 2:45 p.m., 17 October 1991
'17OCT1991:12:5'DT	12:05 p.m., 17 October 1991
'17OCT1991:2:0'DT	2:00 a.m., 17 October 1991
'17OCT1991:0:0'DT	Midnight, 17 October 1991

SAS Time Values

The SAS System also supports *time values*. SAS time values are just like datetime values, except that the date part is not given. To write a time value in a SAS program, write the time the same as for a datetime constant, but use T instead of DT. For example, 2:45:32 p.m. is written '14:45:32'T. Time values are represented by a number of seconds since midnight, so SAS reads '14:45:32'T as 53132.

SAS time values are not very useful for identifying time series, since usually both the date and the time of day are needed. Time values are not discussed further in this book.

Reading Date and Datetime Values with Informats

SAS provides a selection of *informats* for reading SAS date and datetime values from date and time values recorded in ordinary notations.

A SAS informat is an instruction that converts the values from a character-string representation into the internal numerical value of a SAS variable. Date informats convert dates from ordinary notations used to enter them to SAS date values; datetime informats convert date and time from ordinary notation to SAS datetime values.

For example, the following SAS statements read monthly values of the U.S. Consumer Price Index. Since the data are monthly, you could identify the date with the variables YEAR and MONTH, as in the previous example. Instead, in this example the time periods are coded as a three-letter month abbreviation followed by the year. The informat MONYY. is used to read month-year dates coded this way and to express them as SAS date values for the first day of the month, as follows:

```
data uscpi;
  input date : monyy7. cpi;
  format date monyy7.;
  label cpi = "US Consumer Price Index";
datalines;
jun1990 129.9
jul1990 130.4
aug1990 131.6

... more lines ...
```

The SAS System provides informats for most common notations for dates and times. For more information about the date and datetime informats available, see [Chapter 4](#).

Formatting Date and Datetime Values

SAS provides *formats* to convert the internal representation of date and datetime values used by SAS to ordinary notations for dates and times. Several different formats are available for displaying dates and datetime values in most of the commonly used notations.

A SAS format is an instruction that converts the internal numerical value of a SAS variable to a character string that can be printed or displayed. Date formats convert SAS date values to a readable form; datetime formats convert SAS datetime values to a readable form.

In the preceding example, the variable DATE was set to the SAS date value for the first day of the month for each observation. If the data set USCPI were printed or otherwise displayed, the values shown for DATE would be the number of days since 1 January 1960. (See the “DATE with no format” column in [Figure 3.2](#).) To display date values appropriately, use the FORMAT statement.

The following example processes the data set USCPI to make several copies of the variable DATE and uses a FORMAT statement to give different formats to these copies. The format cases shown are the MONYY7. format (for the DATE variable), the DATE9. format (for the DATE1 variable), and no format (for the DATE0 variable). The PROC PRINT output in [Figure 3.2](#) shows the effect of the different formats on how the date values are printed.

```

data fmttest;
  set uscpi;
  date0 = date;
  date1 = date;
  label date = "DATE with MONYY7. format"
        date1 = "DATE with DATE9. format"
        date0 = "DATE with no format";
  format date monyy7. date1 date9.;
run;

proc print data=fmttest label;
run;

```

Figure 3.2 SAS Date Values Printed with Different Formats

Obs	DATE with MONYY7. format	US Consumer Price Index	DATE with no format	DATE with DATE9. format
1	JUN1990	129.9	11109	01JUN1990
2	JUL1990	130.4	11139	01JUL1990
3	AUG1990	131.6	11170	01AUG1990
4	SEP1990	132.7	11201	01SEP1990
5	OCT1990	133.5	11231	01OCT1990
6	NOV1990	133.8	11262	01NOV1990
7	DEC1990	133.8	11292	01DEC1990
8	JAN1991	134.6	11323	01JAN1991
9	FEB1991	134.8	11354	01FEB1991
10	MAR1991	135.0	11382	01MAR1991
11	APR1991	135.2	11413	01APR1991
12	MAY1991	135.6	11443	01MAY1991
13	JUN1991	136.0	11474	01JUN1991
14	JUL1991	136.2	11504	01JUL1991

The appropriate format to use for SAS date or datetime valued ID variables depends on the sampling frequency or periodicity of the time series. [Table 3.2](#) shows recommended formats for common data sampling frequencies and shows how the date '17OCT1991'D or the datetime value '17OCT1991:14:45:32'DT is displayed by these formats.

Table 3.2 Formats for Different Sampling Frequencies

ID Values	Periodicity	FORMAT	Example
SAS date	Annual	YEAR4.	1991
	Quarterly	YYQC6.	1991:4
	Monthly	MONYY7.	OCT1991
	Weekly	WEEKDATX23.	Thursday, 17 Oct 1991
	Daily	DATE9.	17OCT1991
SAS datetime	Hourly	DATETIME10.	17OCT91:14
	Minutes	DATETIME13.	17OCT91:14:45
	Seconds	DATETIME16.	17OCT91:14:45:32

For more information about the date and datetime formats available, see Chapter 4, “[Date Intervals, Formats, and Functions](#).”

The Variables DATE and DATETIME

SAS/ETS procedures enable you to identify time series observations in many different ways to suit your needs. As discussed in preceding sections, you can use a combination of several ID variables, such as YEAR and MONTH for monthly data.

However, using a single SAS date or datetime ID variable is more convenient and enables you to take advantage of some features SAS/ETS procedures provide for processing ID variables. One such feature is automatic extrapolation of the ID variable to identify forecast observations. These features are discussed in the following sections.

Thus, it is a good practice to include a SAS date or datetime ID variable in all the time series SAS data sets you create. It is also a good practice to always give the date or datetime ID variable a format appropriate for the data periodicity. (For information about creating SAS date and datetime values from multiple ID variables, see the section “[Computing Dates from Calendar Variables](#)” on page 94.)

You can assign a SAS date- or datetime-valued ID variable any name that conforms to SAS variable name requirements. However, you might find working with time series data in SAS easier and less confusing if you adopt the practice of always using the same name for the SAS date or datetime ID variable.

This book always names the date- or datetime-values ID variable DATE if it contains SAS date values or DATETIME if it contains SAS datetime values. This makes it easy to recognize the ID variable and also makes it easy to recognize whether this ID variable uses SAS date or datetime values.

Sorting by Time

Many SAS/ETS procedures assume the data are in chronological order. If the data are not in time order, you can use the SORT procedure to sort the data set. For example:

```
proc sort data=a;
  by date;
run;
```

There are many ways of coding the time ID variable or variables, and some ways do not sort correctly. If you use SAS date or datetime ID values as suggested in the preceding section, you do not need to be concerned with this issue. But if you encode date values in nonstandard ways, you need to consider whether your ID variables will sort.

SAS date and datetime values always sort correctly, as do combinations of numeric variables such as YEAR, MONTH, and DAY used together. Julian dates also sort correctly. (Julian dates are numbers of the form *yyddd*, where *yy* is the year and *ddd* is the day of the year. For example, 17 October 1991 has the Julian date value 91290.)

Calendar dates such as numeric values coded as *mmddy* or *ddmmyy* do not sort correctly. Character variables that contain display values of dates, such as dates in the notation produced by SAS date formats, generally do not sort correctly.

Subsetting Data and Selecting Observations

It is often necessary to subset data for analysis. You might need to subset data to do the following:

- restrict the time range. For example, you want to perform a time series analysis using only recent data and ignoring observations from the distant past.
- select cross sections of the data. (See the section “[Cross-Sectional Dimensions and BY Groups](#)” on page 79.) For example, you have a data set with observations over time for each of several states, and you want to analyze the data for a single state.
- select particular kinds of time series from an interleaved-form data set. (See the section “[Interleaved Time Series](#)” on page 81.) For example, you have an output data set produced by the FORECAST procedure that contains both forecast and confidence limits observations, and you want to extract only the forecast observations.
- exclude particular observations. For example, you have an outlier in your time series, and you want to exclude this observation from the analysis.

You can subset data either by using the DATA step to create a subset data set or by using a WHERE statement with the SAS procedure that analyzes the data.

A typical WHERE statement used in a procedure has the following form:

```
proc arima data=full;  
  where '31dec1993'd < date < '26mar1994'd;  
  identify var=close;  
run;
```

For complete reference documentation on the WHERE statement, see *SAS Language Reference: Dictionary*.

Subsetting SAS Data Sets

To create a subset data set, specify the name of the subset data set in the DATA statement, bring in the full data set with a SET statement, and specify the subsetting criteria with either subsetting IF statements or WHERE statements.

For example, suppose you have a data set that contains time series observations for each of several states. The following DATA step uses a WHERE statement to exclude observations with dates before 1970 and uses a subsetting IF statement to select observations for North Carolina (NC):

```
data subset;  
  set full;  
  where date >= '1jan1970'd;  
  if state = 'NC';  
run;
```

In this case, it makes no difference logically whether the WHERE statement or the IF statement is used, and you can combine several conditions in one subsetting statement. The following statements produce the same results as the previous example:

```
data subset;
  set full;
  if date >= '1jan1970'd & state = 'NC';
run;
```

The WHERE statement acts on the input data sets specified in the SET statement before observations are processed by the DATA step program, whereas the IF statement is executed as part of the DATA step program. If the input data set is indexed, using the WHERE statement can be more efficient than using the IF statement. However, the WHERE statement can refer only to variables in the input data set, not to variables computed by the DATA step program.

To subset the variables of a data set, use KEEP or DROP statements or use KEEP= or DROP= data set options. For more information about KEEP and DROP statements and SAS data set options, see *SAS Language Reference: Dictionary*.

For example, suppose you want to subset the data set as in the preceding example, but you want to include in the subset data set only the variables DATE, X, and Y. You could use the following statements:

```
data subset;
  set full;
  if date >= '1jan1970'd & state = 'NC';
  keep date x y;
run;
```

Using the WHERE Statement with SAS Procedures

Use the WHERE statement with SAS procedures to process only a subset of the input data set. For example, suppose you have a data set that contains monthly observations for each of several states, and you want to use the AUTOREG procedure to analyze data since 1970 for North Carolina (NC). You could use the following statements:

```
proc autoreg data=full;
  where date >= '1jan1970'd & state = 'NC';
  ... additional statements ...
run;
```

You can specify any number of conditions in the WHERE statement. For example, suppose that a strike created an outlier in May 1975, and you want to exclude that observation. You could use the following statements:

```
proc autoreg data=full;
  where date >= '1jan1970'd & state = 'NC'
    & date ^= '1may1975'd;
  ... additional statements ...
run;
```

Using SAS Data Set Options

You can use the OBS= and FIRSTOBS= data set options to subset the input data set.

For example, the following statements print observations 20 through 25 of the data set FULL:

```
proc print data=full(firstobs=20 obs=25);
run;
```

Figure 3.3 Partial Listing of Data Set FULL

Obs	date	state	i	x	y	close
20	21OCT1993	NC	20	0.44803	0.35302	0.44803
21	22OCT1993	NC	21	0.03186	1.67414	0.03186
22	23OCT1993	NC	22	-0.25232	-1.61289	-0.25232
23	24OCT1993	NC	23	0.42524	0.73112	0.42524
24	25OCT1993	NC	24	0.05494	-0.88664	0.05494
25	26OCT1993	NC	25	-0.29096	-1.17275	-0.29096

You can use KEEP= and DROP= data set options to exclude variables from the input data set. For information about SAS data set options, see *SAS Language Reference: Dictionary*.

Storing Time Series in a SAS Data Set

This section discusses aspects of storing time series in SAS data sets. The topics discussed are the standard form of a time series data set, storing several series with different time ranges in the same data set, omitted observations, cross-sectional dimensions and BY groups, and interleaved time series.

Any number of time series can be stored in a SAS data set. Normally, each time series is stored in a separate variable. For example, the following statements augment the US CPI data set read in the previous example with values for the producer price index (PPI):

```
data usprice;
  input date : monyy7. cpi ppi;
  format date monyy7.;
  label cpi = "Consumer Price Index"
        ppi = "Producer Price Index";
datalines;
jun1990 129.9 114.3
jul1990 130.4 114.5
aug1990 131.6 116.5

... more lines ...

proc print data=usprice;
run;
```

Figure 3.4 Time Series Data Set Containing Two Series

Obs	date	cpi	ppi
1	JUN1990	129.9	114.3
2	JUL1990	130.4	114.5
3	AUG1990	131.6	116.5
4	SEP1990	132.7	118.4
5	OCT1990	133.5	120.8
6	NOV1990	133.8	120.1
7	DEC1990	133.8	118.7
8	JAN1991	134.6	119.0
9	FEB1991	134.8	117.2
10	MAR1991	135.0	116.2
11	APR1991	135.2	116.0
12	MAY1991	135.6	116.5
13	JUN1991	136.0	116.3
14	JUL1991	136.2	116.0

Standard Form of a Time Series Data Set

The simple way the CPI and PPI time series are stored in the USPRICE data set in the preceding example is termed the *standard form* of a time series data set. A time series data set in standard form has the following characteristics:

- The data set contains one variable for each time series.
- The data set contains exactly one observation for each time period.
- The data set contains an ID variable or variables that identify the time period of each observation.
- The data set is sorted by the ID variables associated with date time values, so the observations are in time sequence.
- The data are equally spaced in time. That is, successive observations are a fixed time interval apart, so the data set can be described by a single sampling interval such as hourly, daily, monthly, quarterly, yearly, and so forth. This means that time series with different sampling frequencies are not mixed in the same SAS data set.

Most SAS/ETS procedures that process time series expect the input data set to contain time series in this standard form, and this is the simplest way to store time series in SAS data sets. (The [EXPAND](#) and [TIMESERIES](#) procedures can be helpful in converting your data to this standard form.) There are more complex ways to represent time series in SAS data sets.

You can incorporate cross-sectional dimensions with BY groups, so that each BY group is like a standard form time series data set. This method is discussed in the section “[Cross-Sectional Dimensions and BY Groups](#)” on page 79.

You can interleave time series, with several observations for each time period identified by another ID variable. Interleaved time series data sets are used to store several series in the same SAS variable. Interleaved time

series data sets are often used to store series of actual values, predicted values, and residuals, or series of forecast values and confidence limits for the forecasts. This is discussed in the section “[Interleaved Time Series](#)” on page 81.

Several Series with Different Ranges

Different time series can have values recorded over different time ranges. Since a SAS data set must have the same observations for all variables, when time series with different ranges are stored in the same data set, missing values must be used for the periods in which a series is not available.

Suppose that in the previous example you did not record values for CPI before August 1990 and did not record values for PPI after June 1991. The USPRICE data set could be read with the following statements:

```
data usprice;
    input date : monyy7. cpi ppi;
    format date monyy7.;
datalines;
jun1990      . 114.3
jul1990      . 114.5
aug1990 131.6 116.5
sep1990 132.7 118.4
oct1990 133.5 120.8
nov1990 133.8 120.1
dec1990 133.8 118.7
jan1991 134.6 119.0
feb1991 134.8 117.2
mar1991 135.0 116.2
apr1991 135.2 116.0
may1991 135.6 116.5
jun1991 136.0 116.3
jul1991 136.2      .
;
```

The decimal points with no digits in the data records represent missing data and are read by SAS as missing value codes.

In this example, the time range of the USPRICE data set is June 1990 through July 1991, but the time range of the CPI variable is August 1990 through July 1991, and the time range of the PPI variable is June 1990 through June 1991.

SAS/ETS procedures ignore missing values at the beginning or end of a series. That is, the series is considered to begin with the first nonmissing value and end with the last nonmissing value.

Missing Values and Omitted Observations

Missing data can also occur within a series. Missing values that appear after the beginning of a time series and before the end of the time series are called *embedded missing values*.

Suppose that in the preceding example you did not record values for CPI for November 1990 and did not record values for PPI for both November 1990 and March 1991. The USPRICE data set could be read with the following statements:

```

data usprice;
    input date : monyy. cpi ppi;
    format date monyy.;
datalines;
jun1990      . 114.3
jul1990      . 114.5
aug1990 131.6 116.5
sep1990 132.7 118.4
oct1990 133.5 120.8
nov1990      .      .
dec1990 133.8 118.7
jan1991 134.6 119.0
feb1991 134.8 117.2
mar1991 135.0      .
apr1991 135.2 116.0
may1991 135.6 116.5
jun1991 136.0 116.3
jul1991 136.2      .
;

```

In this example, the series CPI has one embedded missing value, and the series PPI has two embedded missing values. The ranges of the two series are the same as before.

Note that the observation for November 1990 has missing values for both CPI and PPI; there are no data for this period. This is an example of a *missing observation*.

You might ask why the data record for this period is included in the example at all, since the data record contains no data. However, deleting the data record for November 1990 from the example would cause an *omitted observation* in the USPRICE data set. SAS/ETS procedures expect input data sets to contain observations for a contiguous time sequence. If you omit observations from a time series data set and then try to analyze the data set with SAS/ETS procedures, the omitted observations will cause errors. When all data are missing for a period, a missing observation should be included in the data set to preserve the time sequence of the series.

If observations are omitted from the data set, the [EXPAND](#) procedure can be used to fill in the gaps with missing values (or to interpolate nonmissing values) for the time series variables and with the appropriate date or datetime values for the ID variable.

Cross-Sectional Dimensions and BY Groups

Often, time series in a collection are related by a cross sectional dimension. For example, the national average U.S. consumer price index data shown in the previous example can be disaggregated to show price indexes for major cities. In this case, there are several related time series: CPI for New York, CPI for Chicago, CPI for Los Angeles, and so forth. When these time series are considered as one data set, the city whose price level is measured is a cross sectional dimension of the data.

There are two basic ways to store such related time series in a SAS data set. The first way is to use a standard form time series data set with a different variable for each series.

For example, the following statements read CPI series for three major U.S. cities:

```
data citycpi;
    input date : monyy7. cpiny cpichi cpila;
    format date monyy7.;
datalines;
nov1989 133.200 126.700 130.000
dec1989 133.300 126.500 130.600
jan1990 135.100 128.100 132.100

... more lines ...
```

The second way is to store the data in a time series cross-sectional form. In this form, the series for all cross sections are stored in one variable and a cross section ID variable is used to identify observations for the different series. The observations are sorted by the cross section ID variable and by time within each cross section.

The following statements indicate how to read the CPI series for U.S. cities in time series cross-sectional form:

```
data cpicity;
    length city $11;
    input city $11. date : monyy. cpi;
    format date monyy.;
datalines;
New York      JAN1990    135.100
New York      FEB1990    135.300
New York      MAR1990    136.600

... more lines ...

proc sort data=cpicity;
    by city date;
run;
```

When processing a time series cross sectional form data set with most SAS/ETS procedures, use the cross section ID variable in a BY statement to process the time series separately. The data set must be sorted by the cross section ID variable and sorted by date within each cross section. The PROC SORT step in the preceding example ensures that the CPICITY data set is correctly sorted.

When the cross section ID variable is used in a BY statement, each BY group in the data set is like a standard form time series data set. Thus, SAS/ETS procedures that expect a standard form time series data set can process time series cross sectional data sets when a BY statement is used, producing an independent analysis for each cross section.

It is also possible to analyze time series cross-sectional data jointly. The [PANEL](#) procedure (and the older [TSCSREG](#) procedure) expects the input data to be in the time series cross-sectional form described here. For more information, see Chapter 26, “[The PANEL Procedure](#).”

Interleaved Time Series

Normally, a time series data set has only one observation for each time period, or one observation for each time period within a cross section for a time series cross-sectional-form data set. However, it is sometimes useful to store several related time series in the same variable when the different series do not correspond to levels of a cross-sectional dimension of the data.

In this case, the different time series can be interleaved. An interleaved time series data set is similar to a time series cross-sectional data set, except that the observations are sorted differently and the ID variable that distinguishes the different time series does not represent a cross-sectional dimension.

Some SAS/ETS procedures produce interleaved output data sets. The interleaved time series form is a convenient way to store procedure output when the results consist of several different kinds of series for each of several input series. (Interleaved time series are also easy to process with plotting procedures. See the section “[Plotting Time Series](#)” on page 86.)

For example, the [FORECAST](#) procedure fits a model to each input time series and computes predicted values and residuals from the model. The FORECAST procedure then uses the model to compute forecast values beyond the range of the input data and also to compute upper and lower confidence limits for the forecast values.

Thus, the output from PROC FORECAST consists of up to five related time series for each variable forecast. The five resulting time series for each input series are stored in a single output variable with the same name as the series that is being forecast. The observations for the five resulting series are identified by values of the variable `_TYPE_`. These observations are interleaved in the output data set with observations for the same date grouped together.

The following statements show how to use PROC FORECAST to forecast the variable CPI in the USCPI data set. [Figure 3.5](#) shows part of the output data set produced by PROC FORECAST and illustrates the interleaved structure of this data set.

```
proc forecast data=uscpi interval=month lead=12
              out=foreout outfull outresid;
    var cpi;
    id date;
run;

proc print data=foreout (obs=6);
run;
```

Figure 3.5 Partial Listing of Output Data Set Produced by PROC FORECAST

Obs	date	_TYPE_	_LEAD_	cpi
1	JUN1990	ACTUAL	0	129.900
2	JUN1990	FORECAST	0	130.817
3	JUN1990	RESIDUAL	0	-0.917
4	JUL1990	ACTUAL	0	130.400
5	JUL1990	FORECAST	0	130.678
6	JUL1990	RESIDUAL	0	-0.278

Observations with `_TYPE_=ACTUAL` contain the values of CPI read from the input data set. Observations with `_TYPE_=FORECAST` contain one-step-ahead predicted values for observations with dates in the range of the input series and contain forecast values for observations for dates beyond the range of the input series. Observations with `_TYPE_=RESIDUAL` contain the difference between the actual and one-step-ahead predicted values. Observations with `_TYPE_=U95` and `_TYPE_=L95` contain the upper and lower bounds, respectively, of the 95% confidence interval for the forecasts.

Using Interleaved Data Sets as Input to SAS/ETS Procedures

Interleaved time series data sets are not directly accepted as input by SAS/ETS procedures. However, it is easy to use a `WHERE` statement with any procedure to subset the input data and select one of the interleaved time series as the input.

For example, to analyze the residual series contained in the `PROC FORECAST` output data set with another SAS/ETS procedure, include a `WHERE _TYPE_='RESIDUAL'` statement. The following statements perform a spectral analysis of the residuals produced by `PROC FORECAST` in the preceding example:

```
proc spectra data=foreout out=spectout;
  var cpi;
  where _type_='RESIDUAL';
run;
```

Combined Cross Sections and Interleaved Time Series Data Sets

Interleaved time series output data sets produced from BY-group processing of time series cross-sectional input data sets have a complex structure that combines a cross-sectional dimension, a time dimension, and the values of the `_TYPE_` variable. For example, consider the `PROC FORECAST` output data set produced by the following statements:

```
title "FORECAST Output Data Set with BY Groups";

proc forecast data=cpicity interval=month
  method=expo lead=2
  out=foreout outfull outresid;
  var cpi;
  id date;
  by city;
run;

proc print data=foreout (obs=6);
run;
```

The output data set `FOREOUT` contains many different time series in the single variable `CPI`. (The first few observations of `FOREOUT` are shown in [Figure 3.6](#).) BY groups that are identified by the variable `CITY` contain the result series for the different cities. Within each value of `CITY`, the actual, forecast, residual, and confidence limits series are stored in interleaved form, with the observations for the different series identified by the values of `_TYPE_`.

Figure 3.6 Combined Cross Sections and Interleaved Time Series Data**FORECAST Output Data Set with BY Groups**

Obs	city	date	_TYPE_	_LEAD_	cpi
1	Chicago	JAN90	ACTUAL	0	128.100
2	Chicago	JAN90	FORECAST	0	128.252
3	Chicago	JAN90	RESIDUAL	0	-0.152
4	Chicago	FEB90	ACTUAL	0	129.200
5	Chicago	FEB90	FORECAST	0	128.896
6	Chicago	FEB90	RESIDUAL	0	0.304

Output Data Sets of SAS/ETS Procedures

Some SAS/ETS procedures (such as PROC FORECAST) produce interleaved output data sets, and other SAS/ETS procedures produce standard form time series data sets. The form a procedure uses depends on whether the procedure is normally used to produce multiple result series for each of many input series in one step (as PROC FORECAST does).

For example, the [ARIMA](#) procedure can output actual series, forecast series, residual series, and confidence limit series just as the FORECAST procedure does. The PROC ARIMA output data set uses the standard form because PROC ARIMA is designed for the detailed analysis of one series at a time and so forecasts only one series at a time.

The following statements show the use of the ARIMA procedure to produce a forecast of the USCPI data set. [Figure 3.7](#) shows part of the output data set that is produced by the ARIMA procedure's FORECAST statement. (The printed output from PROC ARIMA is not shown.) Compare the PROC ARIMA output data set shown in [Figure 3.7](#) with the PROC FORECAST output data set shown in [Figure 3.6](#).

```

title "PROC ARIMA Output Data Set";

proc arima data=uscpi;
    identify var=cpi(1);
    estimate q=1;
    forecast id=date interval=month
              lead=12 out=arimaout;

run;

proc print data=arimaout (obs=6);
run;
```

Figure 3.7 Partial Listing of Output Data Set Produced by PROC ARIMA**PROC ARIMA Output Data Set**

Obs	date	cpi	FORECAST	STD	L95	U95	RESIDUAL
1	JUN1990	129.9	.	.	.	.	.
2	JUL1990	130.4	130.368	0.36160	129.660	131.077	0.03168
3	AUG1990	131.6	130.881	0.36160	130.172	131.590	0.71909
4	SEP1990	132.7	132.354	0.36160	131.645	133.063	0.34584
5	OCT1990	133.5	133.306	0.36160	132.597	134.015	0.19421
6	NOV1990	133.8	134.046	0.36160	133.337	134.754	-0.24552

The output data set produced by the ARIMA procedure's FORECAST statement stores the actual values in a variable with the same name as the response series, stores the forecast series in a variable named FORECAST, stores the residuals in a variable named RESIDUAL, stores the 95% confidence limits in variables named L95 and U95, and stores the standard error of the forecast in the variable STD.

This method of storing several different result series as a standard form time series data set is simple and convenient. However, it works well only for a single input series. The forecast of a single series can be stored in the variable FORECAST. But if two series are forecast, two different FORECAST variables are needed.

The STATESPACE procedure handles this problem by generating forecast variable names FOR1, FOR2, and so forth. The SPECTRA procedure uses a similar method. Names such as FOR1, FOR2, RES1, RES2, and so forth require you to remember the order in which the input series are listed. This is why PROC FORECAST, which is designed to forecast a whole list of input series at once, stores its results in interleaved form.

Other SAS/ETS procedures are often used for a single input series but can also be used to process several series in a single step. Thus, they are not clearly like PROC FORECAST nor clearly like PROC ARIMA in the number of input series they are designed to work with. These procedures use a third method for storing multiple result series in an output data set. These procedures store output time series in standard form (as PROC ARIMA does) but require an OUTPUT statement to give names to the result series.

Time Series Periodicity and Time Intervals

A fundamental characteristic of time series data is how frequently the observations are spaced in time. How often the observations of a time series occur is called the *sampling frequency* or the *periodicity* of the series. For example, a time series with one observation each month has a monthly sampling frequency or monthly periodicity and so is called a monthly time series.

In SAS, data periodicity is described by specifying periodic *time intervals* into which the dates of the observations fall. For example, the SAS time interval MONTH divides time into calendar months.

Many SAS/ETS procedures enable you to specify the periodicity of the input data set with the INTERVAL= option. For example, specifying INTERVAL=MONTH indicates that the procedure should expect the ID variable to contain SAS date values, and that the date value for each observation should fall in a separate calendar month. The EXPAND procedure uses interval name values with the FROM= and TO= options to control the interpolation of time series from one periodicity to another.

SAS also uses time intervals in several other ways. In addition to indicating the periodicity of time series data sets, time intervals are used with the interval functions INTNX and INTCK and for controlling the plot axis and reference lines for plots of data over time.

Specifying Time Intervals

Intervals are specified in SAS by using *interval names* such as YEAR, QTR, MONTH, DAY, and so forth. Table 3.3 summarizes the basic types of intervals.

Table 3.3 Basic Interval Types

Name	Periodicity
YEAR	Yearly
SEMIYEAR	Semiannual
QTR	Quarterly
MONTH	Monthly
SEMIMONTH	1st and 16th of each month
TENDAY	1st, 11th, and 21st of each month
WEEK	Weekly
WEEKDAY	Daily, ignoring weekend days
DAY	Daily
HOURLY	Hourly
MINUTE	Every minute
SECOND	Every second

Interval names can be abbreviated in various ways. For example, you could specify monthly intervals as MONTH, MONTHS, MONTHLY, or just MON. SAS accepts all these forms as equivalent.

Interval names can also be qualified with a multiplier to indicate multiperiod intervals. For example, biennial intervals are specified as YEAR2.

Interval names can also be qualified with a shift index to indicate intervals with different starting points. For example, fiscal years starting in July are specified as YEAR.7.

Intervals are classified as either date or datetime intervals. Date intervals are used with SAS date values, while datetime intervals are used with SAS datetime values. The interval types YEAR, SEMIYEAR, QTR, MONTH, SEMIMONTH, TENDAY, WEEK, WEEKDAY, and DAY are date intervals. HOUR, MINUTE, and SECOND are datetime intervals. Date intervals can be turned into datetime intervals for use with datetime values by prefixing the interval name with 'DT'. Thus DTMONTH intervals are like MONTH intervals but are used with datetime ID values instead of date ID values.

For more information about specifying time intervals and for a detailed reference to the different kinds of intervals available, see Chapter 4, “[Date Intervals, Formats, and Functions](#).”

Using Intervals with SAS/ETS Procedures

SAS/ETS procedures use the date or datetime interval and the ID variable in the following ways:

- to validate the data periodicity. The ID variable is used to check the data and verify that successive observations have valid ID values that correspond to successive time intervals.
- to check for gaps in the input observations. For example, if INTERVAL=MONTH and an input observation for January 1990 is followed by an observation for April 1990, there is a gap in the input data with two omitted observations.
- to label forecast observations in the output data set. The values of the ID variable for the forecast observations after the end of the input data set are extrapolated according to the frequency specifications of the INTERVAL= option.

Time Intervals, the Time Series Forecasting System, and the Time Series Viewer

Time intervals are used in the Time Series Forecasting System and Time Series Viewer to identify the number of seasonal cycles or seasonality associated with a DATE, DATETIME, or TIME ID variable. For example, monthly time series have a seasonality of 12 because there are 12 months in a year; quarterly time series have a seasonality of 4 because there are four quarters in a year. The seasonality is used to analyze seasonal properties of time series data and to estimate seasonal forecasting methods.

Plotting Time Series

This section discusses SAS procedures that are available for plotting time series data, but it covers only certain aspects of the use of these procedures with time series data.

The Time Series Viewer displays and analyzes time series plots for time series data sets that do not contain cross sections. See Chapter 56, “[Getting Started with Time Series Forecasting](#).”

The SGPLOT procedure produces high resolution color graphics plots. For more information, see *SAS ODS Graphics: Procedures Guide* and *SAS/GRAPH: Reference*.

Using the Time Series Viewer

The following command starts the Time Series Viewer to display the plot of CPI in the USCPI data set against DATE. (The USCPI data set was shown in the previous example; the time series used in the following example contains more observations than previously shown.)

```
tsvview data=uscpi var=cpi timeid=date
```

The TSVIEW DATA= option specifies the data set to be viewed; the VAR= option specifies the variable that contains the time series observations; the TIMEID= option specifies the time series ID variable.


The Time Series Viewer can also be invoked by selecting **Solutions►Analyze►Time Series Viewer** from the menu in the SAS Display Manager.

Using PROC SGPLOT

The following statements use the SGPLOT procedure to plot CPI in the USCPI data set against DATE. (The USCPI data set was shown in a previous example; the data set plotted in the following example contains more observations than shown previously.)

```
title "Plot of USCPI Data";  
proc sgplot data=uscpi;  
    series x=date y=cpi / markers;  
run;
```

The plot is shown in [Figure 3.8](#).


Figure 3.8 Plot of Monthly CPI over Time

Controlling the Time Axis: Tick Marks and Reference Lines

It is possible to control the spacing of the tick marks on the time axis. The following statements use the XAXIS statement to tell PROC SGPLOT to mark the axis at the start of each quarter:

```
proc sgplot data=uscpi;
  series x=date y=cpi / markers;
  format date yyqc.;
  xaxis values=('1jan90'd to '1jul91'd by qtr);
run;
```

The plot is shown in [Figure 3.9](#).

Figure 3.9 Plot of Monthly CPI over Time

Overlay Plots of Different Variables

You can plot two or more series stored in different variables on the same graph by specifying multiple plot requests in one SGPLOT statement.

For example, the following statements plot the CPI, FORECAST, L95, and U95 variables produced by PROC ARIMA in a previous example. A reference line is drawn to mark the start of the forecast period. Quarterly tick marks with YYQC format date values are used.

```

title "ARIMA Forecasts of CPI";

proc arima data=uscpi;
  identify var=cpi(1);
  estimate q=1;
  forecast id=date interval=month lead=12 out=arimaout;
run;

title "ARIMA forecasts of CPI";
proc sgplot data=arimaout noautolegend;
  scatter x=date y=cpi;

```


```

scatter x=date y=forecast / markerattrs=(symbol=asterisk);
scatter x=date y=l95 / markerattrs=(symbol=asterisk color=green);
scatter x=date y=u95 / markerattrs=(symbol=asterisk color=green);
format date yyqc4.;
xaxis values=('1jan90'd to '1jul92'd by qtr);
refline '15jul91'd / axis=x;
run;

```

The plot is shown in [Figure 3.10](#).

Figure 3.10 Plot of ARIMA Forecast



Overlay Plots of Interleaved Series

You can also plot several series on the same graph when the different series are stored in the same variable in interleaved form. Plot interleaved time series by using the values of the ID variable in `GROUP=` option to distinguish the different series.

The following example plots the output data set produced by PROC FORECAST in a previous example. Since the residual series has a different scale than the other series, it is excluded from the plot with a `WHERE` statement.

The `_TYPE_` variable is used in the PLOT statement to identify the different series and to select the SCATTER statements to use for each plot.

```

title "Plot of Forecasts of US CPI Data";


proc forecast data=uscpi interval=month lead=12
      out=foreout outfull outresid;
      var cpi;
      id date;
run;

proc sgplot data=foreout;
  where _type_ ^= 'RESIDUAL';
  scatter x=date y=cpi / group=_type_ markerattrs=(symbol=asterisk);
  format date yyqc4.;
  xaxis values=('1jan90'd to '1jul92'd by qtr);
  refline '15jul91'd / axis=x;
run;

```

The plot is shown in [Figure 3.11](#).

Figure 3.11 Plot of Forecast




Residual Plots

The following example plots the residuals series that was excluded from the plot in the previous example. The NEEDLE statement specifies a needle plot, so that each residual point is plotted as a vertical line showing deviation from zero.

```
title "Plot of Residuals for USCPI Data";
proc sgplot data=foreout;
  where _type_ = 'RESIDUAL';
  needle x=date y=cpi / markers;
  format date yyqc4.;
  xaxis values=('1jan90'd to '1jul91'd by qtr);
run;
```

The plot is shown in [Figure 3.12](#).

Figure 3.12 Plot of Residuals



Using PROC GPLOT


The GPLOT procedure in SAS/GRAPH software can also be used to plot time series data, although the newer SGPLOT procedure is easier to use.

The following is an example of how PROC GPLOT can be used to produce a plot similar to the graph produced by PROC SGPLOT in the preceding section:

```
title "Plot of USCPI Data";  
proc gplot data=uscpi;  
    symbol i=spline v=circle h=2;  
    plot cpi * date;  
run;
```

The plot is shown in [Figure 3.13](#).

Figure 3.13 Plot of Monthly CPI over Time



For more information about the GPLOT procedure, see *SAS/GRAPH: Reference*.

Calendar and Time Functions

Calendar and time functions convert calendar and time variables such as YEAR, MONTH, DAY, and HOUR, MINUTE, SECOND into SAS date or datetime values, and vice versa.

The SAS calendar and time functions are DATEJUL, DATEPART, DAY, DHMS, HMS, HOUR, JULDATE, MDY, MINUTE, MONTH, QTR, SECOND, TIMEPART, WEEKDAY, YEAR, and YYQ. For more information about these functions, see *SAS Language Reference: Dictionary*.

Computing Dates from Calendar Variables

The MDY function converts MONTH, DAY, and YEAR values to a SAS date value. For example, MDY(2010,17,91) returns the SAS date value '17OCT2010'D.

The YYQ function computes the SAS date for the first day of a quarter. For example, YYQ(2010,4) returns the SAS date value '1OCT2010'D.

The DATEJUL function computes the SAS date for a Julian date. For example, DATEJUL(91290) returns the SAS date '17OCT2010'D.

The YYQ and MDY functions are useful for creating SAS date variables when the ID values recorded in the data are year and quarter; year and month; or year, month, and day.

For example, the following statements read quarterly data from records in which dates are coded as separate year and quarter values. The YYQ function is used to compute the variable DATE.

```
data usecon;
    input year qtr gnp;
    date = yyq( year, qtr );
    format date yyqc.;
datalines;
1990 1 5375.4
1990 2 5443.3
1990 3 5514.6

... more lines ...
```

The monthly US CPI data shown in a previous example contained time ID values represented in the MONYY format. If the data records instead contain separate year and month values, the data can be read in and the DATE variable computed with the following statements:

```
data uscpi;
    input month year cpi;
    date = mdy( month, 1, year );
    format date monyy.;
datalines;
6 90 129.9
7 90 130.4
8 90 131.6

... more lines ...
```

Computing Calendar Variables from Dates

The functions YEAR, MONTH, DAY, WEEKDAY, and JULDATE compute calendar variables from SAS date values.

Returning to the example of reading the USCPI data from records that contain date values represented in the MONYY format, you can find the month and year of each observation from the SAS dates of the observations by using the following statements:

```
data uscpi;
    input date monyy7. cpi;
    format date monyy7.;
    year  = year( date );
    month = month( date );
datalines;
jun1990 129.9
jul1990 130.4
aug1990 131.6
... more lines ...
```

Converting between Date, Datetime, and Time Values

The DATEPART function computes the SAS date value for the date part of a SAS datetime value. The TIMEPART function computes the SAS time value for the time part of a SAS datetime value.

The HMS function computes SAS time values from HOUR, MINUTE, and SECOND time variables. The DHMS function computes a SAS datetime value from a SAS date value and HOUR, MINUTE, and SECOND time variables.

For more information about these functions, see the section “[SAS Date, Time, and Datetime Functions](#)” on page 145.

Computing Datetime Values

To compute datetime ID values from calendar and time variables, first compute the date and then compute the datetime with DHMS.

For example, suppose you read tri-hourly temperature data with time recorded as YEAR, MONTH, DAY, and HOUR. The following statements show how to compute the ID variable DATETIME:

```
data weather;
    input year month day hour temp;
    datetime = dhms( mdy( month, day, year ), hour, 0, 0 );
    format datetime datetime10.;
datalines;
91 10 16 21 61
91 10 17 0 56
```

```

91 10 17 3 53
91 10 17 6 54

... more lines ...

```

Computing Calendar and Time Variables

The functions HOUR, MINUTE, and SECOND compute time variables from SAS datetime values. The DATEPART function and the date-to-calendar variables functions can be combined to compute calendar variables from datetime values.

For example, suppose the date and time of the tri-hourly temperature data in the preceding example were recorded as datetime values in the datetime format. The following statements show how to compute the YEAR, MONTH, DAY, and HOUR of each observation and include these variables in the SAS data set:

```

data weather;
    input datetime : datetime13. temp;
    format datetime datetime10.;
    hour = hour( datetime );
    date = datepart( datetime );
    year = year( date );
    month = month( date );
    day = day( date );
datalines;
16oct91:21:00 61
17oct91:00:00 56
17oct91:03:00 53
17oct91:06:00 54

... more lines ...

```

Interval Functions INTNX and INTCK

The SAS interval functions INTNX and INTCK perform calculations with date values, datetime values, and time intervals. They can be used for calendar calculations with SAS date values to increment date values or datetime values by intervals and to count time intervals between dates.

The INTNX function increments dates by intervals. INTNX computes the date or datetime of the start of the interval a specified number of intervals from the interval that contains a given date or datetime value.

The form of the INTNX function is

INTNX (*interval*, *from*, *n* < , *alignment* >);

The arguments to the INTNX function are as follows:

interval

is a character constant or variable that contains an interval name

from

is a SAS date value (for date intervals) or datetime value (for datetime intervals)

n

is the number of intervals to increment from the interval that contains the *from* value

alignment

controls the alignment of SAS dates, within the interval, used to identify output observations. Allowed values are BEGINNING, MIDDLE, END, and SAMEDAY.

The number of intervals to increment, *n*, can be positive, negative, or zero.

For example, the statement `NEXTMON=INTNX('MONTH',DATE,1)` assigns to the variable `NEXTMON` the date of the first day of the month following the month that contains the value of `DATE`. Thus `INTNX('MONTH','21OCT2007'D,1)` returns the date 1 November 2007.

The `INTCK` function counts the number of interval boundaries between two date values or between two datetime values.

The form of the `INTCK` function is

INTCK (*interval*, *from*, *to*) ;

The arguments of the `INTCK` function are as follows:

interval

is a character constant or variable that contains an interval name.

from

is the starting date value (for date intervals) or datetime value (for datetime intervals).

to

is the ending date value (for date intervals) or datetime value (for datetime intervals).

For example, the statement `NEWYEARS=INTCK('YEAR',DATE1,DATE2)` assigns to the variable `NEWYEARS` the number of New Year's Days between the two dates.

Incrementing Dates by Intervals

Use the `INTNX` function to increment dates by intervals. For example, suppose you want to know the date of the start of the week that is six weeks from the week of 17 October 1991. The function `INTNX('WEEK','17OCT91'D,6)` returns the SAS date value '24NOV1991'D.

One practical use of the `INTNX` function is to generate periodic date values. For example, suppose the monthly U.S. Consumer Price Index data in a previous example were recorded without any time identifier on the data records. Given that you know the first observation is for June 1990, the following statements use the `INTNX` function to compute the `ID` variable `DATE` for each observation:

```

data uscpi;
    input cpi;
    date = intnx( 'month', '1jun1990'd, _n_-1 );
    format date monyy7.;
datalines;
129.9
130.4
131.6

... more lines ...

```

The automatic variable `_N_` counts the number of times the DATA step program has executed; in this case `_N_` contains the observation number. Thus `_N_-1` is the increment needed from the first observation date. Alternatively, you could increment from the month before the first observation, in which case the INTNX function in this example would be written `INTNX('MONTH','1MAY1990'D,_N_)`.

Alignment of SAS Dates

Any date within the time interval that corresponds to an observation of a periodic time series can serve as an ID value for the observation. For example, the USCPI data in a previous example might have been recorded with dates at the 15th of each month. The person recording the data might reason that since the CPI values are monthly averages, midpoints of the months might be the appropriate ID values.

However, as far as SAS/ETS procedures are concerned, what is important about monthly data is the month of each observation, not the exact date of the ID value. If you indicate that the data are monthly (with an `INTERVAL=MONTH`) option, SAS/ETS procedures ignore the day of the month in processing the ID variable. The `MONYY` format also ignores the day of the month.

Thus, you could read in the monthly USCPI data with mid-month DATE values by using the following statements:

```

data uscpi;
    input date : date9. cpi;
    format date monyy7.;
datalines;
15jun1990 129.9
15jul1990 130.4
15aug1990 131.6

... more lines ...

```

The results of using this version of the USCPI data set for analysis with SAS/ETS procedures would be the same as with first-of-month values for DATE. Although you can use any date within the interval as an ID value for the interval, you might find working with time series in SAS less confusing if you always use date ID values normalized to the start of the interval.

For some applications it might be preferable to use end of period dates, such as 31Jan1994, 28Feb1994, 31Mar1994, ..., 31Dec1994. For other applications, such as plotting time series, it might be more convenient to use interval midpoint dates to identify the observations.

(Some SAS/ETS procedures provide an `ALIGN=` option to control the alignment of dates for output time series observations. In addition, the `INTNX` library function supports an optional argument to specify the alignment of the returned date value.)

To normalize date values to the start of intervals, use the `INTNX` function with a 0 increment. The `INTNX` function with an increment of 0 computes the date of the first day of the interval (or the first second of the interval for datetime values).

For example, `INTNX('MONTH','17OCT1991'D,0,'BEG')` returns the date '1OCT1991'D.

The following statements show how the preceding example can be changed to normalize the mid-month `DATE` values to first-of-month and end-of-month values. For exposition, the first-of-month value is transformed back into a middle-of-month value.

```
data uscpi;
  input date : date9. cpi;
  format date monyy7.;
  monthbeg = intnx( 'month', date, 0, 'beg' );
  midmonth = intnx( 'month', monthbeg, 0, 'mid' );
  monthend = intnx( 'month', date, 0, 'end' );
datalines;
15jun1990 129.9
15jul1990 130.4
15aug1990 131.6

... more lines ...
```

If you want to compute the date of a particular day within an interval, you can use calendar functions, or you can increment the starting date of the interval by a number of days. The following example shows three ways to compute the seventh day of the month:

```
data test;
  set uscpi;
  mon07_1 = mdy( month(date), 7, year(date) );
  mon07_2 = intnx( 'month', date, 0, 'beg' ) + 6;
  mon07_3 = intnx( 'day', date, 6 );
run;
```

Computing the Width of a Time Interval

To compute the width of a time interval, subtract the ID value of the start of the next interval from the ID value of the start of the current interval. If the ID values are SAS dates, the width is in days. If the ID values are SAS datetime values, the width is in seconds.

For example, the following statements show how to add a variable `WIDTH` to the `USCPI` data set that contains the number of days in the month for each observation:

```
data uscpi;
  input date : date9. cpi;
  format date monyy7.;
  width = intnx( 'month', date, 1 ) - intnx( 'month', date, 0 );
datalines;
15jun1990 129.9
15jul1990 130.4
```

```
15aug1990 131.6
15sep1990 132.7

... more lines ...
```

Computing the Ceiling of an Interval

To shift a date to the start of the next interval if it is not already at the start of an interval, subtract 1 from the date and use INTNX to increment the date by 1 interval.

For example, the following statements add the variable NEWYEAR to the monthly USCPI data set. The variable NEWYEAR contains the date of the next New Year's Day. NEWYEAR contains the same value as DATE when the DATE value is the start of year and otherwise contains the date of the start of the next year.

```
data test;
  set uscpi;
  newyear = intnx( 'year', date - 1, 1 );
  format newyear date.;
run;
```

Counting Time Intervals

Use the INTCK function to count the number of interval boundaries between two dates.

Note that the INTCK function counts the number of times the beginning of an interval is reached in moving from the first date to the second. It does not count the number of complete intervals between two dates. Following are two examples:

- The function INTCK('MONTH', '1JAN1991'D, '31JAN1991'D) returns 0, since the two dates are within the same month.
- The function INTCK('MONTH', '31JAN1991'D, '1FEB1991'D) returns 1, since the two dates lie in different months that are one month apart.

When the first date is later than the second date, INTCK returns a negative count. For example, the function INTCK('MONTH', '1FEB1991'D, '31JAN1991'D) returns -1.

The following example shows how to use the INTCK function with shifted interval specifications to count the number of Sundays, Mondays, Tuesdays, and so forth, in each month. The variables NSUNDAY, NMONDAY, NTUESDAY, and so forth, are added to the USCPI data set.

```
data uscpi;
  set uscpi;
  d0 = intnx( 'month', date, 0 ) - 1;
  d1 = intnx( 'month', date, 1 ) - 1;
  nSunday  = intck( 'week.1', d0, d1 );
  nMonday  = intck( 'week.2', d0, d1 );
  nTuesday = intck( 'week.3', d0, d1 );
```

```

nWednesday = intck( 'week.4', d0, d1 );
nThursday  = intck( 'week.5', d0, d1 );
nFriday    = intck( 'week.6', d0, d1 );
nSaturday  = intck( 'week.7', d0, d1 );
drop d0 d1;
run;

```

Since the INTCK function counts the number of interval beginning dates between two dates, the number of Sundays is computed by counting the number of week boundaries between the last day of the previous month and the last day of the current month. To count Mondays, Tuesdays, and so forth, shifted week intervals are used. The interval type WEEK.2 specifies weekly intervals starting on Mondays, WEEK.3 specifies weeks starting on Tuesdays, and so forth.

Checking Data Periodicity

Suppose you have a time series data set and you want to verify that the data periodicity is correct, the observations are dated correctly, and the data set is sorted by date. You can use the INTCK function to compare the date of the current observation with the date of the previous observation and verify that the dates fall into consecutive time intervals.

For example, the following statements verify that the data set US CPI is a correctly dated monthly data set. The RETAIN statement is used to hold the date of the previous observation, and the automatic variable `_N_` is used to start the verification process with the second observation.

```

data _null_;
  set uscpi;
  retain prevdate;
  if _n_ > 1 then
    if intck( 'month', prevdate, date ) ^= 1 then
      put "Bad date sequence at observation number " _n_;
  prevdate = date;
run;

```

Filling In Omitted Observations in a Time Series Data Set

Most SAS/ETS procedures expect input data to be in the standard form, with no omitted observations in the sequence of time periods. When data are missing for a time period, the data set should contain a missing observation, in which all variables except the ID variables have missing values.

You can replace omitted observations in a time series data set with missing observations with the [EXPAND](#) procedure.

The following statements create a monthly data set, OMITTED, from data lines that contain records for an intermittent sample of months. (Data values are not shown.) The OMITTED data set is sorted to make sure it is in time order.

```

data omitted;
  input date : monyy7. x y z;
  format date monyy7.;
datalines;
jan1991 ...

```

```

mar1991  ...
apr1991  ...
jun1991  ...
... etc. ...
;

proc sort data=omitted;
    by date;
run;

```

This data set is converted to a standard form time series data set by the following PROC EXPAND step. The TO= option specifies that monthly data is to be output, while the METHOD=NONE option specifies that no interpolation is to be performed, so that the variables X, Y, and Z in the output data set STANDARD will have missing values for the omitted time periods that are filled in by the EXPAND procedure.

```

proc expand data=omitted
            out=standard
            to=month
            method=none;
    id date;
run;

```

Using Interval Functions for Calendar Calculations

With a little thought, you can come up with a formula that involves INTNX and INTCK functions and different interval types to perform almost any calendar calculation.

For example, suppose you want to know the date of the third Wednesday in the month of October 1991. The answer can be computed as

```
intnx( 'week.4', '1oct91'd - 1, 3 )
```

which returns the SAS date value '16OCT91'D.

Consider this more complex example: how many weekdays are there between 17 October 1991 and the second Friday in November 1991, inclusive? The following formula computes the number of weekdays between the date value contained in the variable DATE and the second Friday of the following month (including the ending dates of this period):

```

n = intck( 'weekday', date - 1,
          intnx( 'week.6', intnx( 'month', date, 1 ) - 1, 2 ) + 1 );

```

Setting DATE to '17OCT91'D and applying this formula produces the answer, N=17.

Lags, Leads, Differences, and Summations

When working with time series data, you sometimes need to refer to the values of a series in previous or future periods. For example, the usual interest in the consumer price index series shown in previous examples is how fast the index is changing, rather than the actual level of the index. To compute a percent change, you

need both the current and the previous values of the series. When you model a time series, you might want to use the previous values of other series as explanatory variables.

This section discusses how to use the DATA step to perform operations over time: lags, differences, leads, summations over time, and percent changes.

The EXPAND procedure can also be used to perform many of these operations; for more information, see Chapter 15, “[The EXPAND Procedure](#).” See also the section “[Transforming Time Series](#)” on page 111.

The LAG and DIF Functions

The DATA step provides two functions, LAG and DIF, for accessing previous values of a variable or expression. These functions are useful for computing lags and differences of series.

For example, the following statements add the variables CPILAG and CPIDIF to the USCPI data set. The variable CPILAG contains lagged values of the CPI series. The variable CPIDIF contains the changes of the CPI series from the previous period; that is, CPIDIF is CPI minus CPILAG. The new data set is shown in part in [Figure 3.14](#).

```
data uscpi;
    set uscpi;
    cpilag = lag( cpi );
    cpidif = dif( cpi );
run;

proc print data=uscpi;
run;
```

Figure 3.14 USCPI Data Set with Lagged and Differenced Series

Plot of USCPI Data

Obs	date	cpi	cpilag	cpidif
1	JUN1990	129.9	.	.
2	JUL1990	130.4	129.9	0.5
3	AUG1990	131.6	130.4	1.2
4	SEP1990	132.7	131.6	1.1
5	OCT1990	133.5	132.7	0.8
6	NOV1990	133.8	133.5	0.3
7	DEC1990	133.8	133.8	0.0
8	JAN1991	134.6	133.8	0.8
9	FEB1991	134.8	134.6	0.2
10	MAR1991	135.0	134.8	0.2
11	APR1991	135.2	135.0	0.2
12	MAY1991	135.6	135.2	0.4
13	JUN1991	136.0	135.6	0.4
14	JUL1991	136.2	136.0	0.2

Understanding the DATA Step LAG and DIF Functions

When used in this simple way, LAG and DIF act as lag and difference functions. However, it is important to keep in mind that, despite their names, the LAG and DIF functions available in the DATA step are not true lag and difference functions.

Rather, LAG and DIF are queuing functions that remember and return argument values from previous calls. The LAG function remembers the value you pass to it and returns as its result the value you passed to it on the previous call. The DIF function works the same way but returns the difference between the current argument and the remembered value. (LAG and DIF return a missing value the first time the function is called.)

A true lag function does not return the value of the argument for the “previous call,” as do the DATA step LAG and DIF functions. Instead, a true lag function returns the value of its argument for the “previous observation,” regardless of the sequence of previous calls to the function. Thus, for a true lag function to be possible, it must be clear what the “previous observation” is.

If the data are sorted chronologically, then LAG and DIF act as true lag and difference functions. If in doubt, use PROC SORT to sort your data before using the LAG and DIF functions. Beware of missing observations, which can cause LAG and DIF to return values that are not the actual lag and difference values.

The DATA step is a powerful tool that can read any number of observations from any number of input files or data sets, can create any number of output data sets, and can write any number of output observations to any of the output data sets, all in the same program. Thus, in general, it is not clear what “previous observation” means in a DATA step program. In a DATA step program, the “previous observation” exists only if you write the program in a simple way that makes this concept meaningful.

Since, in general, the previous observation is not clearly defined, it is not possible to make true lag or difference functions for the DATA step. Instead, the DATA step provides queuing functions that make it easy to compute lags and differences.

Pitfalls of DATA Step LAG and DIF Functions

The LAG and DIF functions compute lags and differences provided that the sequence of calls to the function corresponds to the sequence of observations in the output data set. However, any complexity in the DATA step that breaks this correspondence causes the LAG and DIF functions to produce unexpected results.

For example, suppose you want to add the variable CPILAG to the USCPI data set, as in the previous example, and you also want to subset the series to 1991 and later years. You might use the following statements:

```
data subset;
  set uscpi;
  if date >= '1jan1991'd;
  cpilag = lag( cpi ); /* WRONG PLACEMENT! */
run;
```

If the subsetting IF statement comes before the LAG function call, the value of CPILAG will be missing for January 1991, even though a value for December 1990 is available in the USCPI data set. To avoid losing this value, you must rearrange the statements to ensure that the LAG function is actually executed for the December 1990 observation.

```
data subset;
  set uscpi;
  cpilag = lag( cpi );
  if date >= '1jan1991'd;
run;
```

In other cases, the subsetting statement should come before the LAG and DIF functions. For example, the following statements subset the FOREOUT data set shown in a previous example to select only `_TYPE_=RESIDUAL` observations and also to compute the variable `LAGRESID`:

```
data residual;
  set foreout;
  if _type_ = "RESIDUAL";
  lagresid = lag( cpi );
run;
```

Another pitfall of LAG and DIF functions arises when they are used to process time series cross-sectional data sets. For example, suppose you want to add the variable `CPILAG` to the `CPICITY` data set shown in a previous example. You might use the following statements:

```
data cpicity;
  set cpicity;
  cpilag = lag( cpi );
run;
```

However, these statements do not yield the desired result. In the data set produced by these statements, the value of `CPILAG` for the first observation for the first city is missing (as it should be), but in the first observation for all later cities, `CPILAG` contains the last value for the previous city. To correct this, set the lagged variable to missing at the start of each cross section, as follows:

```
data cpicity;
  set cpicity;
  by city date;
  cpilag = lag( cpi );
  if first.city then cpilag = .;
run;
```

Alternatives to LAG and DIF Functions

You can also use the [EXPAND](#) procedure to compute lags and differences. For example, the following statements compute lag and difference variables for `CPI`:

```
proc expand data=uscpi out=uscpi method=none;
  id date;
  convert cpi=cpilag / transform=( lag 1 );
  convert cpi=cpidif / transform=( dif 1 );
run;
```

You can also calculate lags and differences in the `DATA` step without using `LAG` and `DIF` functions. For example, the following statements add the variables `CPILAG` and `CPIDIF` to the `USCPI` data set:

```
data uscpi;
  set uscpi;
  retain cpilag;
  cpidif = cpi - cpilag;
  output;
  cpilag = cpi;
run;
```

The `RETAIN` statement prevents the `DATA` step from reinitializing `CPILAG` to a missing value at the start of each iteration and thus allows `CPILAG` to retain the value of `CPI` assigned to it in the last statement.

The OUTPUT statement causes the output observation to contain values of the variables before CPILAG is reassigned the current value of CPI in the last statement. This is the approach that must be used if you want to build a variable that is a function of its previous lags.

LAG and DIF Functions in PROC MODEL

The preceding discussion of LAG and DIF functions applies to LAG and DIF functions available in the DATA step. However, LAG and DIF functions are also used in the MODEL procedure.

The MODEL procedure LAG and DIF functions do not work like the DATA step LAG and DIF functions. The LAG and DIF functions supported by PROC MODEL are true lag and difference functions, not queuing functions.

Unlike the DATA step, the MODEL procedure processes observations from a single input data set, so the “previous observation” is always clearly defined in a PROC MODEL program. Therefore, PROC MODEL is able to define LAG and DIF as true lagging functions that operate on values from the previous observation. For more information about LAG and DIF functions in the MODEL procedure, see Chapter 25, “[The MODEL Procedure](#).”

Multiperiod Lags and Higher-Order Differencing

To compute lags at a lagging period greater than 1, add the lag length to the end of the LAG keyword to specify the lagging function needed. For example, the LAG2 function returns the value of its argument two calls ago, the LAG3 function returns the value of its argument three calls ago, and so forth.

To compute differences at a lagging period greater than 1, add the lag length to the end of the DIF keyword. For example, the DIF2 function computes the differences between the value of its argument and the value of its argument two calls ago. (The maximum lagging period is 100.)

The following statements add the variables CPILAG12 and CPIDIF12 to the USCPI data set. CPILAG12 contains the value of CPI from the same month one year ago. CPIDIF12 contains the change in CPI from the same month one year ago. (In this case, the first 12 values of CPILAG12 and CPIDIF12 are missing.)

```
data uscpi;
  set uscpi;
  cpilag12 = lag12( cpi );
  cpidif12 = dif12( cpi );
run;
```

To compute second differences, take the difference of the difference. To compute higher-order differences, nest DIF functions to the order needed. For example, the following statements compute the second difference of CPI:

```
data uscpi;
  set uscpi;
  cpi2dif = dif( dif( cpi ) );
run;
```

Multiperiod lags and higher-order differencing can be combined. For example, the following statements compute monthly changes in the inflation rate, with inflation rate computed as percent change in CPI from the same month one year ago:

```
data uscpi;
  set uscpi;
  infchnng = dif( 100 * dif12( cpi ) / lag12( cpi ) );
run;
```

Percent Change Calculations

There are several common ways to compute the percent change in a time series. This section illustrates the use of LAG and DIF functions by showing SAS statements for various kinds of percent change calculations.

Computing Period-to-Period Change

To compute percent change from the previous period, divide the difference of the series by the lagged value of the series and multiply by 100.

```
data uscpi;
  set uscpi;
  pctchnng = dif( cpi ) / lag( cpi ) * 100;
  label pctchnng = "Monthly Percent Change, At Monthly Rates";
run;
```

Often, changes from the previous period are expressed at annual rates. This is done by exponentiation of the current-to-previous period ratio to the number of periods in a year and expressing the result as a percent change. For example, the following statements compute the month-over-month change in CPI as a percent change at annual rates:

```
data uscpi;
  set uscpi;
  pctchnng = ( ( cpi / lag( cpi ) ) ** 12 - 1 ) * 100;
  label pctchnng = "Monthly Percent Change, At Annual Rates";
run;
```

Computing Year-over-Year Change

To compute percent change from the same period in the previous year, use LAG and DIF functions with a lagging period equal to the number of periods in a year. (For quarterly data, use LAG4 and DIF4. For monthly data, use LAG12 and DIF12.)

For example, the following statements compute monthly percent change in CPI from the same month one year ago:

```
data uscpi;
  set uscpi;
  pctchnng = dif12( cpi ) / lag12( cpi ) * 100;
  label pctchnng = "Percent Change from One Year Ago";
run;
```

To compute year-over-year percent change measured at a given period within the year, subset the series of percent changes from the same period in the previous year to form a yearly data set. Use an IF or WHERE statement to select observations for the period within each year on which the year-over-year changes are based.

For example, the following statements compute year-over-year percent change in CPI from December of the previous year to December of the current year:

```
data annual;
  set uscpi;
  pctchnng = dif12( cpi ) / lag12( cpi ) * 100;
  label pctchnng = "Percent Change: December to December";
  if month( date ) = 12;
  format date year4.;
run;
```

Computing Percent Change in Yearly Averages

To compute changes in yearly averages, first aggregate the series to an annual series by using the EXPAND procedure, and then compute the percent change of the annual series. (For more information about PROC EXPAND, see Chapter 15, “[The EXPAND Procedure](#).”)

For example, the following statements compute percent changes in the annual averages of CPI:

```
proc expand data=uscpi out=annual from=month to=year;
  convert cpi / observed=average method=aggregate;
run;

data annual;
  set annual;
  pctchnng = dif( cpi ) / lag( cpi ) * 100;
  label pctchnng = "Percent Change in Yearly Averages";
run;
```

It is also possible to compute percent change in the average over the most recent yearly span. For example, the following statements compute monthly percent change in the average of CPI over the most recent 12 months from the average over the previous 12 months:

```
data uscpi;
  retain sum12 0;
  drop sum12 ave12 cpilag12;
  set uscpi;
  sum12 = sum12 + cpi;
  cpilag12 = lag12( cpi );
  if cpilag12 ^= . then sum12 = sum12 - cpilag12;
  if lag11( cpi ) ^= . then ave12 = sum12 / 12;
  pctchnng = dif12( ave12 ) / lag12( ave12 ) * 100;
  label pctchnng = "Percent Change in 12 Month Moving Ave.";
run;
```

This example is a complex use of LAG and DIF functions that requires care in handling the initialization of the moving-window averaging process. The LAG12 of CPI is checked for missing values to determine when more than 12 values have been accumulated, and older values must be removed from the moving sum. The LAG11 of CPI is checked for missing values to determine when at least 12 values have been accumulated; AVE12 will be missing when LAG11 of CPI is missing. The DROP statement prevents temporary variables from being added to the data set.

Note that the DIF and LAG functions must execute for every observation, or the queues of remembered values will not operate correctly. The CPILAG12 calculation must be separate from the IF statement. The PCTCHNG calculation must not be conditional on the IF statement.

The EXPAND procedure provides an alternative way to compute moving averages.

Leading Series

Although the SAS System does not provide a function to look ahead at the “next” value of a series, there are a couple of ways to perform this task.

The most direct way to compute leads is to use the EXPAND procedure. For example:

```
proc expand data=uscpi out=uscpi method=none;
  id date;
  convert cpi=cpilead1 / transform=( lead 1 );
  convert cpi=cpilead2 / transform=( lead 2 );
run;
```

Another way to compute lead series in SAS software is by lagging the time ID variable, renaming the series, and merging the result data set back with the original data set.

For example, the following statements add the variable CPILEAD to the USCPI data set. The variable CPILEAD contains the value of CPI in the following month. (The value of CPILEAD is missing for the last observation, of course.)

```
data temp;
  set uscpi;
  keep date cpi;
  rename cpi = cpilead;
  date = lag( date );
  if date ^= .;
run;

data uscpi;
  merge uscpi temp;
  by date;
run;
```

To compute leads at different lead lengths, you must create one temporary data set for each lead length. For example, the following statements compute CPILEAD1 and CPILEAD2, which contain leads of CPI for 1 and 2 periods, respectively:

```
data temp1(rename=(cpi=cpilead1))
  temp2(rename=(cpi=cpilead2));
  set uscpi;
  keep date cpi;
  date = lag( date );
  if date ^= . then output temp1;
  date = lag( date );
  if date ^= . then output temp2;
run;

data uscpi;
  merge uscpi temp1 temp2;
  by date;
run;
```

Summing Series

Simple cumulative sums are easy to compute using SAS sum statements. The following statements show how to compute the running sum of variable X in data set A, adding XSUM to the data set:

```
data a;
  set a;
  xsum + x;
run;
```

The SAS sum statement automatically retains the variable XSUM and initializes it to 0, and the sum statement treats missing values as 0. The sum statement is equivalent to using a RETAIN statement and the SUM function. The previous example could also be written as follows:

```
data a;
  set a;
  retain xsum;
  xsum = sum( xsum, x );
run;
```

You can also use the EXPAND procedure to compute summations. For example:

```
proc expand data=a out=a method=none;
  convert x=xsum / transform=( sum );
run;
```

Like differencing, summation can be done at different lags and can be repeated to produce higher-order sums. To compute sums over observations separated by lags greater than 1, use the LAG and SUM functions together, and use a RETAIN statement that initializes the summation variable to zero.

For example, the following statements add the variable XSUM2 to data set A. XSUM2 contains the sum of every other observation, with even-numbered observations containing a cumulative sum of values of X from even observations, and odd-numbered observations containing a cumulative sum of values of X from odd observations.

```
data a;
  set a;
  retain xsum2 0;
  xsum2 = sum( lag( xsum2 ), x );
run;
```

Assuming that A is a quarterly data set, the following statements compute running sums of X for each quarter. XSUM4 contains the cumulative sum of X for all observations for the same quarter as the current quarter. Thus, for a first-quarter observation, XSUM4 contains a cumulative sum of current and past first-quarter values.

```
data a;
  set a;
  retain xsum4 0;
  xsum4 = sum( lag3( xsum4 ), x );
run;
```

To compute higher-order sums, repeat the preceding process and sum the summation variable. For example, the following statements compute the first and second summations of X:

```
data a;
  set a;
  xsum + x;
  x2sum + xsum;
run;
```

The following statements compute the second order four-period sum of X:

```
data a;
  set a;
  retain xsum4 x2sum4 0;
  xsum4 = sum( lag3( xsum4 ), x );
  x2sum4 = sum( lag3( x2sum4 ), xsum4 );
run;
```

You can also use PROC EXPAND to compute cumulative statistics and moving window statistics. For more information, see Chapter 15, “[The EXPAND Procedure](#).”

Transforming Time Series


It is often useful to transform time series for analysis or forecasting. Many time series analysis and forecasting methods are most appropriate for time series with an unrestricted range, a linear trend, and a constant variance. Series that do not conform to these assumptions can often be transformed to series for which the methods are appropriate.

Transformations can be useful for the following:

- range restrictions. Many time series cannot have negative values or can be limited to a maximum possible value. You can often create a transformed series with an unbounded range.
- nonlinear trends. Many economic time series grow exponentially. Exponential growth corresponds to linear growth in the logarithms of the series.
- series variability that changes over time. Various transformations can be used to stabilize the variance.
- nonstationarity. The %DFTEST macro can be used to test a series for nonstationarity which can then be removed by differencing.

Log Transformation


The logarithmic transformation is often useful for series that must be greater than zero and that grow exponentially. For example, [Figure 3.15](#) shows a plot of an airline passenger miles series. Notice that the series has exponential growth and the variability of the series increases over time. Airline passenger miles must also be zero or greater.

Figure 3.15 Airline Series

The following statements compute the logarithms of the airline series:

```
data lair;  
  set sashelp.air;  
  logair = log( air );  
run;
```

Figure 3.16 shows a plot of the log-transformed airline series. Notice that the log series has a linear trend and constant variance.

Figure 3.16 Log Airline Series

The %LOGTEST macro can help you decide if a log transformation is appropriate for a series. For more information about the %LOGTEST macro, see Chapter 5, “SAS Macros and Functions.”

Other Transformations

The Box-Cox transformation is a general class of transformations that includes the logarithm as a special case. The %BOXCOXAR macro can be used to find an optimal Box-Cox transformation for a time series. For more information about the %BOXCOXAR macro, see Chapter 5.

The logistic transformation is useful for variables with both an upper and a lower bound, such as market shares. The logistic transformation is useful for proportions, percent values, relative frequencies, or probabilities. The logistic function transforms values between 0 and 1 to values that can range from $-\infty$ to $+\infty$.

For example, the following statements transform the variable SHARE from percent values to an unbounded range:

```
data a;
  set a;
  lshare = log( share / ( 100 - share ) );
run;
```

Many other data transformation can be used. You can create virtually any desired data transformation using DATA step statements.

The EXPAND Procedure and Data Transformations

The EXPAND procedure provides a convenient way to transform series. For example, the following statements add variables for the logarithm of AIR and the logistic of SHARE to data set A:

```
proc expand data=a out=a method=none;
  convert air=logair / transform=( log );
  convert share=lshare / transform=( / 100 logit );
run;
```

For a complete list of transformations supported by PROC EXPAND, see [Table 15.2](#) in Chapter 15, “The EXPAND Procedure.”

Manipulating Time Series Data Sets

This section discusses merging, splitting, and transposing time series data sets and interpolating time series data to a higher or lower sampling frequency.

Splitting and Merging Data Sets

In some cases, you might want to separate several time series that are contained in one data set into different data sets. In other cases, you might want to combine time series from different data sets into one data set.

To split a time series data set into two or more data sets that contain subsets of the series, use a DATA step to create the new data sets and use the KEEP= data set option to control which series are included in each new data set. The following statements split the USPRICE data set shown in a previous example into two data sets, USCPI and USPPI:

```
data uscpi(keep=date cpi)
  usppi(keep=date ppi);
  set usprice;
run;
```

If the series have different time ranges, you can subset the time ranges of the output data sets accordingly. For example, if you know that CPI in USPRICE has the range August 1990 through the end of the data set, while PPI has the range from the beginning of the data set through June 1991, you could write the previous example as follows:

```

data uscpi(keep=date cpi)
  usppi(keep=date ppi);
set usprice;
if date >= '1aug1990'd then output uscpi;
if date <= '1jun1991'd then output usppi;
run;

```

To combine time series from different data sets into one data set, list the data sets to be combined in a MERGE statement and specify the dating variable in a BY statement. The following statements show how to combine the US CPI and US PPI data sets to produce the US PRICE data set. It is important to use the BY DATE statement so that observations are matched by time before merging.

```

data usprice;
  merge uscpi usppi;
  by date;
run;

```

Transposing Data Sets

The TRANSPOSE procedure is used to transpose data sets from one form to another. The TRANSPOSE procedure can transpose variables and observations, or transpose variables and observations within BY groups. This section discusses some applications of the TRANSPOSE procedure relevant to time series data sets. For more information about PROC TRANSPOSE, see *Base SAS Procedures Guide*.

Transposing from Interleaved to Standard Time Series Form

The following statements transpose part of the interleaved-form output data set FOREOUT, produced by PROC FORECAST in a previous example, to a standard form time series data set. To reduce the volume of output produced by the example, a WHERE statement is used to subset the input data set.

Observations with _TYPE_=ACTUAL are stored in the new variable ACTUAL; observations with _TYPE_=FORECAST are stored in the new variable FORECAST; and so forth. Note that the method used in this example works only for a single variable.

```

title "Original Data Set";
proc print data=foreout(obs=10);
  where date > '1may1991'd & date < '1oct1991'd;
run;

proc transpose data=foreout out=trans(drop=_name_);
  var cpi;
  id _type_;
  by date;
  where date > '1may1991'd & date < '1oct1991'd;
run;

title "Transposed Data Set";
proc print data=trans(obs=10);
run;

```

The TRANSPOSE procedure adds the variables `_NAME_` and `_LABEL_` to the output data set. These variables contain the names and labels of the variables that were transposed. In this example, there is only one transposed variable, so `_NAME_` has the value `CPI` for all observations. Thus, `_NAME_` and `_LABEL_` are of no interest and are dropped from the output data set by using the `DROP=` data set option. (If none of the variables transposed have a label, PROC TRANSPOSE does not output the `_LABEL_` variable and the `DROP=_LABEL_` option produces a warning message. You can ignore this message, or you can prevent the message by omitting `_LABEL_` from the `DROP=` list.)

The original and transposed data sets are shown in Figure 3.17 and Figure 3.18. (The observation numbers shown for the original data set reflect the operation of the WHERE statement.)

Figure 3.17 Original Data Sets

Original Data Set

Obs	date	_TYPE_	_LEAD_	cpi
37	JUN1991	ACTUAL	0	136.000
38	JUN1991	FORECAST	0	136.146
39	JUN1991	RESIDUAL	0	-0.146
40	JUL1991	ACTUAL	0	136.200
41	JUL1991	FORECAST	0	136.566
42	JUL1991	RESIDUAL	0	-0.366
43	AUG1991	FORECAST	1	136.856
44	AUG1991	L95	1	135.723
45	AUG1991	U95	1	137.990
46	SEP1991	FORECAST	2	137.443

Figure 3.18 Transposed Data Sets

Transposed Data Set

Obs	date	_LABEL_	ACTUAL	FORECAST	RESIDUAL	L95	U95
1	JUN1991	US Consumer Price Index	136.0	136.146	-0.14616	.	.
2	JUL1991	US Consumer Price Index	136.2	136.566	-0.36635	.	.
3	AUG1991	US Consumer Price Index	.	136.856	.	135.723	137.990
4	SEP1991	US Consumer Price Index	.	137.443	.	136.126	138.761

Transposing Cross-Sectional Dimensions

The following statements transpose the variable `CPI` in the `CPICITY` data set shown in a previous example from time series cross-sectional form to a standard form time series data set. (Only a subset of the data shown in the previous example is used here.) Note that the method shown in this example works only for a single variable.

```

title "Original Data Set";
proc print data=cpicity;
run;

proc sort data=cpicity out=temp;
  by date city;

```

```

run;

proc transpose data=temp out=citycpi(drop=_name_);
  var cpi;
  id city;
  by date;
run;

title "Transposed Data Set";
proc print data=citycpi;
run;

```

The names of the variables in the transposed data sets are taken from the city names in the ID variable CITY. The original and the transposed data sets are shown in [Figure 3.19](#) and [Figure 3.20](#).

Figure 3.19 Original Data Sets

Original Data Set

Obs	city	date	cpi	cpilag
1	Chicago	JAN90	128.1	.
2	Chicago	FEB90	129.2	128.1
3	Chicago	MAR90	129.5	129.2
4	Chicago	APR90	130.4	129.5
5	Chicago	MAY90	130.4	130.4
6	Chicago	JUN90	131.7	130.4
7	Chicago	JUL90	132.0	131.7
8	Los Angeles	JAN90	132.1	.
9	Los Angeles	FEB90	133.6	132.1
10	Los Angeles	MAR90	134.5	133.6
11	Los Angeles	APR90	134.2	134.5
12	Los Angeles	MAY90	134.6	134.2
13	Los Angeles	JUN90	135.0	134.6
14	Los Angeles	JUL90	135.6	135.0
15	New York	JAN90	135.1	.
16	New York	FEB90	135.3	135.1
17	New York	MAR90	136.6	135.3
18	New York	APR90	137.3	136.6
19	New York	MAY90	137.2	137.3
20	New York	JUN90	137.1	137.2
21	New York	JUL90	138.4	137.1

Figure 3.20 Transposed Data Sets**Transposed Data Set**

Obs	date	Chicago	Los_Angeles	New_York
1	JAN90	128.1	132.1	135.1
2	FEB90	129.2	133.6	135.3
3	MAR90	129.5	134.5	136.6
4	APR90	130.4	134.2	137.3
5	MAY90	130.4	134.6	137.2
6	JUN90	131.7	135.0	137.1
7	JUL90	132.0	135.6	138.4

The following statements transpose the CITYCPI data set back to the original form of the CPICITY data set. The variable `_NAME_` is added to the data set to tell PROC TRANSPOSE the name of the variable in which to store the observations in the transposed data set. (If the `(DROP=_NAME_ _LABEL_)` option were omitted from the first PROC TRANSPOSE step, this would not be necessary. PROC TRANSPOSE assumes `ID _NAME_` by default.)

The `NAME=CITY` option in the PROC TRANSPOSE statement causes PROC TRANSPOSE to store the names of the transposed variables in the variable CITY. Because PROC TRANSPOSE recodes the values of the CITY variable to create valid SAS variable names in the transposed data set, the values of the variable CITY in the retransposed data set are not the same as in the original. The retransposed data set is shown in Figure 3.21.

```
data temp;
    set citycpi;
    _name_ = 'CPI';
run;

proc transpose data=temp out=retrans name=city;
    by date;
run;

proc sort data=retrans;
    by city date;
run;

title "Retransposed Data Set";
proc print data=retrans;
run;
```

Figure 3.21 Data Set Transposed Back to Original Form**Retransposed Data Set**

Obs	date	city	CPI
1	JAN90	Chicago	128.1
2	FEB90	Chicago	129.2
3	MAR90	Chicago	129.5
4	APR90	Chicago	130.4
5	MAY90	Chicago	130.4
6	JUN90	Chicago	131.7
7	JUL90	Chicago	132.0
8	JAN90	Los_Angeles	132.1
9	FEB90	Los_Angeles	133.6
10	MAR90	Los_Angeles	134.5
11	APR90	Los_Angeles	134.2
12	MAY90	Los_Angeles	134.6
13	JUN90	Los_Angeles	135.0
14	JUL90	Los_Angeles	135.6
15	JAN90	New_York	135.1
16	FEB90	New_York	135.3
17	MAR90	New_York	136.6
18	APR90	New_York	137.3
19	MAY90	New_York	137.2
20	JUN90	New_York	137.1
21	JUL90	New_York	138.4

Time Series Interpolation

The EXPAND procedure interpolates time series. This section provides a brief summary of the use of PROC EXPAND for different kinds of time series interpolation problems. Most of the issues discussed in this section are explained in greater detail in [Chapter 15](#).

By default, the EXPAND procedure performs interpolation by first fitting cubic spline curves to the available data and then computing needed interpolating values from the fitted spline curves. Other interpolation methods can be requested.

Note that interpolating values of a time series does not add any real information to the data because the interpolation process is not the same process that generated the other (nonmissing) values in the series. While time series interpolation can sometimes be useful, great care is needed in analyzing time series that contain interpolated values.

Interpolating Missing Values

To use the EXPAND procedure to interpolate missing values in a time series, specify the input and output data sets in the PROC EXPAND statement, and specify the time ID variable in an ID statement. For example, the

following statements cause PROC EXPAND to interpolate values for missing values of all numeric variables in the data set USPRICE:

```
proc expand data=usprice out=interpl;
    id date;
run;
```

Interpolated values are computed only for embedded missing values in the input time series. Missing values before or after the range of a series are ignored by the EXPAND procedure.

In the preceding example, PROC EXPAND assumes that all series are measured at points in time given by the value of the ID variable. In fact, the series in the USPRICE data set are monthly averages. PROC EXPAND can produce a better interpolation if this is taken into account. The following example uses the FROM=MONTH option to tell PROC EXPAND that the series is monthly and uses the CONVERT statement with the OBSERVED=AVERAGE to specify that the series values are averages over each month:

```
proc expand data=usprice out=interpl
           from=month;
    id date;
    convert cpi ppi / observed=average;
run;
```

Interpolating to a Higher or Lower Frequency

You can use PROC EXPAND to interpolate values of time series at a higher or lower sampling frequency than the input time series. To change the periodicity of time series, specify the time interval of the input data set with the FROM= option, and specify the time interval for the desired output frequency with the TO= option. For example, the following statements compute interpolated weekly values of the monthly CPI and PPI series:

```
proc expand data=usprice out=interpl
           from=month to=week;
    id date;
    convert cpi ppi / observed=average;
run;
```

Interpolating between Stocks and Flows, Levels and Rates

A distinction is made between variables that are measured at points in time and variables that represent totals or averages over an interval. Point-in-time values are often called *stocks* or *levels*. Variables that represent totals or averages over an interval are often called *flows* or *rates*.

For example, the annual series Gross National Product represents the final goods production of over the year and also the yearly average rate of that production. However, the monthly variable Inventory represents the cost of a stock of goods at the end of the month.

The EXPAND procedure can convert between point-in-time values and period average or total values. To convert observation characteristics, specify the input and output characteristics with the OBSERVED= option in the CONVERT statement. For example, the following statements use the monthly average price index values in USPRICE to compute interpolated estimates of the price index levels at the midpoint of each month:

```
proc expand data=usprice out=midpoint
    from=month;
    id date;
    convert cpi ppi / observed=(average,middle);
run;
```

Reading Time Series Data

Time series data can be coded in many different ways. The SAS System can read time series data recorded in almost any form. Earlier sections of this chapter show how to read time series data coded in several commonly used ways. This section shows how to read time series data from data records coded in two other commonly used ways not previously introduced.

Several time series databases distributed by major data vendors can be read into SAS data sets by the DATASOURCE procedure. For more information, see Chapter 12, “[The DATASOURCE Procedure](#).”

The SASECRSP, SASEFAME, and SASEHAVR interface engines enable SAS users to access and process time series data in CRSPAccess data files, FAME databases, and Haver Analytics Data Link Express (DLX) databases, respectively. For more information, see Chapter 46, “[The SASECRSP Interface Engine](#),” Chapter 47, “[The SASEFAME Interface Engine](#),” and Chapter 49, “[The SASEHAVR Interface Engine](#).”

Reading a Simple List of Values

Time series data can be coded as a simple list of values without dating information and with an arbitrary number of observations on each data record. In this case, the INPUT statement must use the trailing “@@” option to retain the current data record after reading the values for each observation, and the time ID variable must be generated with programming statements.

For example, the following statements read the USPRICE data set from data records that contain pairs of values for CPI and PPI. This example assumes you know that the first pair of values is for June 1990.

```
data usprice;
    input cpi ppi @@;
    date = intnx( 'month', '1jun1990'd, _n_-1 );
    format date monyy7.;
datalines;
129.9 114.3 130.4 114.5 131.6 116.5
132.7 118.4 133.5 120.8 133.8 120.1 133.8 118.7
134.6 119.0 134.8 117.2 135.0 116.2 135.2 116.0
135.6 116.5 136.0 116.3 136.2 116.0
;
```

Reading Fully Described Time Series in Transposed Form

Data for several time series can be coded with separate groups of records for each time series. Data files coded this way are transposed from the form required by SAS procedures. Time series data can also be coded with descriptive information about the series included with the data records.

The following example reads time series data for the USPRICE data set coded with separate groups of records for each series. The data records for each series consist of a series description record and one or more value records. The series description record gives the series name, starting month and year of the series, number of values in the series, and a series label. The value records contain the observations of the time series.

The data are first read into a temporary data set that contains one observation for each value of each series.

```
data temp;
  length _name_ $8 _label_ $40;
  keep _name_ _label_ date value;
  format date monyy.;
  input _name_ month year nval _label_ &;
  date = mdy( month, 1, year );
  do i = 1 to nval;
    input value @;
    output;
    date = intnx( 'month', date, 1 );
  end;
datalines;
cpi      8 90 12  Consumer Price Index
131.6 132.7 133.5 133.8 133.8 134.6 134.8 135.0
135.2 135.6 136.0 136.2
ppi      6 90 13  Producer Price Index
114.3 114.5 116.5 118.4 120.8 120.1 118.7 119.0
117.2 116.2 116.0 116.5 116.3
;
```

The following statements sort the data set by date and series name, and the TRANSPOSE procedure is used to transpose the data into a standard form time series data set:

```
proc sort data=temp;
  by date _name_;
run;

proc transpose data=temp out=usprice(drop=_name_);
  by date;
  var value;
run;

proc contents data=usprice;
run;

proc print data=usprice;
run;
```

The final data set is shown in [Figure 3.23](#).

Figure 3.22 Contents of USPRICE Data Set**Retransposed Data Set****The CONTENTS Procedure**

Alphabetic List of Variables and Attributes				
#	Variable	Type	Len	Format Label
3	cpi	Num	8	Consumer Price Index
1	date	Num	8	MONYY.
2	ppi	Num	8	Producer Price Index

Figure 3.23 Listing of USPRICE Data Set**Retransposed Data Set**

Obs	date	ppi	cpi
1	JUN90	114.3	.
2	JUL90	114.5	.
3	AUG90	116.5	131.6
4	SEP90	118.4	132.7
5	OCT90	120.8	133.5
6	NOV90	120.1	133.8
7	DEC90	118.7	133.8
8	JAN91	119.0	134.6
9	FEB91	117.2	134.8
10	MAR91	116.2	135.0
11	APR91	116.0	135.2
12	MAY91	116.5	135.6
13	JUN91	116.3	136.0
14	JUL91	.	136.2

Subject Index

- BY groups
 - cross-sectional dimensions and, 80
- calendar calculations
 - functions for, 94
 - INTCX function and, 102
 - interval functions and, 102
 - INTNX function and, 102
 - time intervals and, 102
- calendar functions
 - date values and, 94
- calendar variables, 94
 - computing dates from, 94
 - computing from dates, 95
 - computing from datetime values, 96
- changing by interpolation
 - frequency, 120
 - periodicity, 120
 - sampling frequency, 120
- changing periodicity
 - EXPAND procedure, 120
 - time series data, 120
- checking data periodicity
 - INTNX function, 101
 - time intervals, 101
- combining time series data sets, 115
- computing ceiling of intervals
 - INTNX function, 100
- computing ending date of intervals
 - INTNX function, 99
- computing lags
 - RETAIN statement, 105
- computing midpoint dates of intervals
 - INTNX function, 99
- computing widths of intervals
 - INTNX function, 99
- counting time intervals
 - INTCK function, 100
- cross-sectional dimensions, 79
 - combined with interleaved time series, 82
 - ID variables for, 80
 - represented by different series, 79
 - represented with BY groups, 80
 - transposing time series, 116
- DATE
 - ID variables, 73
- date values
 - calendar functions and, 94, 95
 - computing datetime values from, 95
 - computing from datetime values, 95
 - difference between dates, 99
 - formats, 71
 - incrementing by intervals, 97
 - informats, 71
 - INTNX function and, 97
 - normalizing to intervals, 99
 - SAS representation for, 69
 - syntax for, 70
 - time intervals and, 98
- DATE variable, 73
- DATETIME
 - ID variables, 73
- datetime values
 - computing calendar variables from, 96
 - computing from calendar variables, 95
 - computing from time variables, 95
 - computing time variables from, 96
 - formats, 71
 - informats, 71
 - SAS representation for, 70
 - syntax for, 70
- DATETIME variable, 73
- DIF function
 - alternatives to, 105
 - differencing and, 103
 - explained, 104
 - higher-order differences, 106
 - introduced, 103
 - MODEL procedure version, 106
 - multiplier lags and, 106
 - percent change calculations and, 107, 108
 - pitfalls of, 104
 - second difference, 106
- difference between dates
 - date values, 99
- differencing
 - DIF function and, 103, 104
 - higher-order, 106
 - MODEL procedure and, 106
 - multiplier differences, 106
 - percent change calculations and, 107, 108
 - RETAIN statement and, 105
 - second difference, 106
 - time series data, 103–108
- embedded missing values, 78

- ending dates
 - of time intervals, 99
- EXPAND procedure
 - changing periodicity, 120
 - discussed, 119
 - interpolation and, 119
 - interpolation of missing values, 119
- FORECAST procedure
 - interleaved time series and, 81
- formats
 - date values, 71
 - datetime values, 71
 - recommended for time series ID, 72
- frequency
 - changing by interpolation, 120
 - of time series observations, 84, 120
 - time intervals and, 84, 120
- higher-order differences
 - DIF function, 106
- higher-order sums
 - summation, 111
- ID variables
 - DATE, 73
 - DATETIME, 73
 - for interleaved time series, 81
 - for time series data, 69
 - sorting by, 73
- incrementing by intervals
 - date values, 97
- incrementing dates
 - by time intervals, 97
 - INTNX function, 97
- informat
 - date values, 71
 - datetime values, 71
- INTCK function
 - calendar calculations and, 102
 - counting time intervals, 100
 - defined, 97
 - time intervals and, 97
- interleaved form
 - output data sets, 83
 - time series data set, 81
- interleaved time series
 - and `_TYPE_` variable, 81, 82
 - combined with cross-sectional dimension, 82
 - defined, 81
 - FORECAST procedure and, 81, 82
 - ID variables for, 81
 - overlay plots of, 90
 - plots of, 90
- interpolation
 - between levels and rates, 120
 - between stocks and flows, 120
 - EXPAND procedure and, 119
 - of missing values, 119
 - time series data, 120
 - to higher frequency, 120
 - to lower frequency, 120
- interpolation of missing values
 - EXPAND procedure, 119
- interval functions, *see* time intervals, functions
- INTERVAL= option
 - time intervals and, 84
- intervals, *see* time intervals
- INTNX function
 - calendar calculations and, 102
 - checking data periodicity, 101
 - computing ceiling of intervals, 100
 - computing ending date of intervals, 99
 - computing midpoint dates of intervals, 99
 - computing widths of intervals, 99
 - date values and, 97
 - defined, 96
 - incrementing dates, 97
 - normalizing dates in intervals, 99
 - time intervals and, 96
- LAG function
 - alternatives to, 105
 - explained, 104
 - introduced, 103
 - MODEL procedure version, 106
 - multi-period lags and, 106
 - percent change calculations and, 107, 108
 - pitfalls of, 104
- lagging
 - time series data, 103–108
- lags
 - LAG function and, 103, 104
 - MODEL procedure and, 106
 - multi-period lagging, 106
 - percent change calculations and, 107, 108
 - RETAIN statement and, 105
- leads
 - calculation of, 109
 - multi-period, 109
 - time series data, 109
- merging series
 - time series data, 115
- merging time series data sets, 115
- midpoint dates
 - of time intervals, 99
- missing observations
 - contrasted with omitted observations, 79

- missing values
 - contrasted with omitted observations, 79
 - embedded in time series, 78
 - interpolation of, 119
 - time series data and, 78
- MODEL procedure
 - differencing and, 106
 - lags and, 106
- MODEL procedure version
 - DIF function, 106
 - LAG function, 106
- moving averages
 - percent change calculations, 108
- multiperiod differences
 - differencing, 106
- multiperiod lagging
 - lags, 106
- multiperiod lags
 - DIF function and, 106
 - LAG function and, 106
 - summation and, 110, 111
- normalizing dates in intervals
 - INTNX function, 99
- normalizing to intervals
 - date values, 99
- omitted observations
 - contrasted with missing values, 79
 - defined, 79
 - in time series data, 79
 - replacing with missing values, 101
- output data sets
 - and the OUTPUT statement, 84
 - different forms of, 83
 - in standard form, 84
 - interleaved form, 83
 - produced by SAS/ETS procedures, 83
- OUTPUT statement
 - SAS/ETS procedures using, 84
- overlay plots
 - of interleaved time series, 90
 - of time series, 89
 - _TYPE_ variable and, 90
- percent change calculations
 - at annual rates, 107
 - DIF function and, 107
 - differencing and, 107
 - introduced, 107
 - LAG function and, 107
 - lags and, 107
 - moving averages, 108
 - period-to-period, 107
 - time series data, 107, 108
 - year-over-year, 107
 - yearly averages, 108
- percent change calculations and
 - DIF function, 108
 - differencing, 108
 - LAG function, 108
 - lags, 108
- periodicity
 - changing by interpolation, 120
 - of time series observations, 72, 84, 120
- periodicity of time series
 - time intervals, 84, 120
- plot axis
 - time intervals and, 88
- plot axis for time series
 - SGPLOT procedure, 88
- plot reference lines
 - time intervals and, 88
- plotting
 - residual, 92
 - time series data, 86
- plotting time series
 - SGPLOT procedure, 87
 - Time Series Viewer procedure, 87
- reference
 - SGPLOT procedure, 88
- replacing with missing values
 - omitted observations, 101
- residual
 - plotting, 92
- RETAIN statement
 - computing lags, 105
 - differencing and, 105
 - lags and, 105
- sampling frequency
 - changing by interpolation, 120
 - of time series, 72, 84, 120
 - time intervals and, 84
- sampling frequency of time series
 - time intervals, 84, 120
- SAS data sets
 - time series data and, 67
- SAS language features
 - for time series data, 66
- second difference
 - DIF function, 106
 - differencing, 106
- SGPLOT procedure
 - plot axis for time series, 88
 - plotting time series, 87
 - reference, 88
 - time series data, 87

- sorting by ID variables, 73
- sorting time series data, 73
- splitting series
 - time series data, 114
- splitting time series data sets, 114
- standard form
 - of time series data set, 77
- starting dates
 - of time intervals, 99
- summation
 - higher-order sums, 111
 - multiperiod lags and, 110, 111
 - of time series, 110
- time functions, 94
- time intervals
 - calendar calculations and, 102
 - ceiling of, 100
 - checking data periodicity, 101
 - counting, 97, 100
 - date values and, 98
 - ending dates of, 99
 - EXPAND procedure and, 120
 - functions for, 96
 - ID values for, 98
 - incrementing dates by, 96, 97
 - INTCK function and, 97, 100
 - INTERVAL= option and, 84
 - intervals, 85
 - INTNX function and, 96
 - midpoint dates of, 99
 - naming, 85
 - periodicity of time series, 84, 120
 - plot axis and, 88
 - plot reference lines and, 88
 - sampling frequency of time series, 84, 120
 - starting dates of, 98, 99
 - syntax for, 85
 - use with SAS/ETS procedures, 86
 - widths of, 99
- time intervals, functions
 - interval functions, 96
- time ranges
 - of time series, 78
- time series cross-sectional form
 - BY groups and, 80
 - ID variables for, 80
 - of time series data set, 80
 - TSCSREG procedure and, 80
- time series data
 - changing periodicity, 120
 - differencing, 103–108
 - embedded missing values in, 78
 - giving dates to, 69
 - ID variable for, 69
 - interpolation, 120
 - interpolation of, 119, 120
 - lagging, 103–108
 - leads, 109
 - merging series, 115
 - missing values and, 78
 - omitted observations in, 79
 - overlay plot of, 89
 - percent change calculations, 107, 108
 - periodicity of, 84, 120
 - plotting, 86
 - reading, 68, 122
 - reading, with DATA step, 121, 122
 - sampling frequency of, 84, 120
 - SAS data sets and, 67
 - SAS language features for, 66
 - SGPLOT procedure, 87
 - sorting, 73
 - splitting series, 114
 - standard form of, 77
 - stored in SAS data sets, 76
 - summation of, 110, 111
 - time ranges of, 78
 - Time Series Viewer, 87
 - transposing, 115, 116
- time series data set
 - interleaved form of, 81
 - time series cross-sectional form of, 80
- Time Series Viewer
 - time series data, 87
- Time Series Viewer procedure
 - plotting time series, 87
- time values
 - defined, 70
 - syntax for, 70
- time variables
 - computing from datetime values, 96
 - introduced, 94
- TRANSPOSE procedure, 115, 116, 118, 122
 - transposing time series and, 115
- transposing
 - time series data, 115, 116
- transposing time series
 - cross-sectional dimensions, 116
 - from interleaved form, 115
 - from standard form, 118
 - to standard form, 115, 116
 - TRANSPOSE procedure and, 115
- TSCSREG procedure
 - time series cross-sectional form and, 80
- _TYPE_ variable
 - and interleaved time series, 81, 82
 - overlay plots, 90

- year-over-year
 - percent change calculations, [107](#)
- yearly averages
 - percent change calculations, [108](#)

Syntax Index

- DATA step
 - DROP statement, [75](#)
 - IF statement, [74](#)
 - KEEP statement, [75](#)
 - WHERE statement, [74](#)
- DATEJUL
 - function, [94](#)
- DATEPART function, [95](#)
- DAY
 - function, [95](#)
- DHMS
 - function, [95](#)
- DIF
 - function, [103](#)
- DIF function
 - MODEL procedure, [106](#)
- HMS
 - function, [95](#)
- INTCK
 - function, [97](#)
- INTNX
 - function, [96](#)
- JULDATE function, [95](#)
- LAG
 - function, [103](#)
- LAG function
 - MODEL procedure, [106](#)
- MDY
 - function, [94](#)
- MONTH
 - function, [95](#)
- PLOT
 - HAXIS=, [88](#)
- TIMEPART function, [95](#)
- TRANSPOSE procedure, [115](#)
- WEEKDAY
 - function, [95](#)
- YEAR
 - function, [95](#)
- YYQ
 - function, [94](#)